

Programming Language Fundamentals (PLaF) Notes

v0.1095 – February 27, 2023

Eduardo Bonelli

Contents

Preface	7
1 A Calculator	9
1.1 Syntax	9
1.1.1 Concrete Syntax	9
1.1.2 Abstract Syntax	10
1.2 Interpreter	11
1.2.1 Specification	11
1.2.2 Implementation	13
1.2.2.1 Implementation: Final	15
1.3 Exercises	17
2 Simple Functional Languages	19
2.1 LET	19
2.1.1 Concrete Syntax	19
2.1.2 Abstract Syntax	19
2.1.3 Environments	20
2.1.4 Interpreter	20
2.1.4.1 Specification	20
2.1.4.2 Implementation: Attempt I	21
2.1.4.3 Weaving Environments	24
2.1.4.4 Implementation: Final	27
2.1.5 Inspecting the Environment	30
2.2 Exercises	31
2.3 PROC	34
2.3.1 Concrete Syntax	35
2.3.2 Abstract Syntax	35
2.3.3 Interpreter	36
2.3.3.1 Specification	36
2.3.3.2 Implementation	36
2.3.4 Dynamic Scoping	38
2.4 Exercises	39
2.5 REC	41
2.5.1 Concrete Syntax	42
2.5.2 Abstract Syntax	42

2.5.3	Interpreter	43
2.5.3.1	Specification	43
2.5.3.2	Implementation	44
3	Imperative Programming	49
3.1	Mutable Data Structures in OCaml	49
3.1.1	A counter object	49
3.1.2	A stack object	49
3.2	EXPLICIT-REFS	50
3.2.1	Concrete Syntax	50
3.2.2	Abstract Syntax	51
3.2.3	Interpreter	51
3.2.3.1	Specification	51
3.2.3.2	Implementing Stores	53
3.2.3.3	Implementation	54
3.2.4	Extended Example: Encoding Objects	56
3.3	IMPLICIT-REFS	56
3.3.1	Concrete Syntax	56
3.3.2	Abstract Syntax	57
3.3.3	Interpreter	58
3.3.3.1	Specification	58
3.3.3.2	Implementation	58
3.3.3.3	letrec Revisited	60
3.4	Parameter Passing Methods	61
3.4.1	Call-by-Value	61
3.4.2	Call-by-Reference	62
3.4.2.1	Modifying the Interpreter	62
3.4.3	Call-by-Name	63
3.4.4	Call-by-Need	64
3.5	Exercises	65
4	Types	69
4.1	CHECKED	69
4.1.1	Concrete Syntax	69
4.1.2	Abstract Syntax	69
4.1.3	Type-Checker	70
4.1.3.1	Specification	70
4.1.3.2	Towards and Implementation	71
4.1.3.3	Implementation	72
4.1.4	Adding Letrec	74
4.1.5	Exercises	74
5	Simple Object-Oriented Language	79
5.1	SOOL	79
5.1.1	Concrete Syntax	79
5.1.2	Abstract Syntax	80
5.1.3	Interpreter	80

5.1.3.1	Specification	80
5.1.3.2	Implementation	80
6	Modules	81
6.1	Syntax	81
6.1.1	Concrete Syntax	81
6.1.2	Abstract Syntax	82
6.2	Interpreter	82
6.2.1	Specification	82
6.2.2	Implementation	83
6.3	Type-Checking	85
6.3.1	Specification	85
6.3.2	Implementation	86
6.4	Further Reading	87
A	Supporting Files	89
A.1	File Structure	89
A.2	Running the Interpreters	91
B	Solution to Selected Exercises	93

Preface

These course notes provide supporting material for CS496 and CS510. They draw from ideas in [Fri08] but have been evolving over the semesters.

Chapter 1

A Calculator

We introduce a toy language called ARITH. In ARITH programs are simple arithmetic expressions. The objective of this chapter is to provide a gentle introduction to various concepts we will be developing later in these notes. These concepts include the syntax of a language and how its programs are executed.

1.1 Syntax

This section presents the syntax of ARITH. The syntax determines what sequences of symbols which make up our code counts as a syntactically correct program. The syntax is typically presented in the form of a grammar and referred to as the concrete syntax. There are many details in the concrete syntax that are irrelevant for executing a program. For example, an expression such as $4 / 2$ might denote a program that divides 4 by 2. But one may also use an expression such as $4 \text{ div } 2$ to denote the same program. Which of these two is considered syntactically correct is determined by the concrete syntax. In order to execute the program, all we need to know is that 4 is being divided by 2, regardless of how the language requires you to write the division operator itself. The abstract syntax of a language is the underlying representation of a syntactically correct program, once we abstract away any inessential, concrete details. It typically takes the form of a tree and is referred to as an Abstract Syntax Tree. A program called a parser, receives a sequence of symbols and determines whether it conforms to the rules of the concrete syntax. If it doesn't it reports a "syntax error"; if it does, it produces an abstract syntax tree. We next address these topics in further detail for ARITH.

1.1.1 Concrete Syntax

The grammar below specifies the concrete syntax of ARITH. It determines what expressions are syntactically correct ARITH programs. Each line is called a **production**. Expressions enclosed in angle brackets are called **non-terminals**. The grammar below only has two non-terminals, $\langle \text{Expression} \rangle$ and $\langle \text{Number} \rangle$. Among all non-terminals one singles out the so called **start non-terminal**. In our case, the start non-terminal is $\langle \text{Expression} \rangle$. Symbols that appear to the right of " $::=$ " and that are not non-terminals are called **terminals**. The grammar below has the following set of terminals: $\{-, /, (,)\}$. Note that we have not specified what terminals are generated by

the $\langle \text{Number} \rangle$ non-terminal. Such non-terminals are known as tokens and are specified outside the grammar, typically by means of regular expressions; the sequence of symbols matching the regular expressions are known as lexemes. In our particular example, the sequence of symbols identified as $\langle \text{Number} \rangle$ shall be either a sequence of digits (e.g. 123) or a sequence of digits prefixed with a “-” and surrounded by parenthesis (e.g. (-123)).

$$\begin{aligned}\langle \text{Expression} \rangle &::= \langle \text{Number} \rangle \\ \langle \text{Expression} \rangle &::= \langle \text{Expression} \rangle - \langle \text{Expression} \rangle \\ \langle \text{Expression} \rangle &::= \langle \text{Expression} \rangle / \langle \text{Expression} \rangle \\ \langle \text{Expression} \rangle &::= (\langle \text{Expression} \rangle)\end{aligned}$$

For the sake of simplicity, our language ARITH only supports two arithmetic operations, subtraction and division. We will later add others.

Examples of syntactically correct expressions are 3-4, ((4/0)-4), 3-4-1, and (-4)/2. That each of these sequence of terminals is a syntactically correct expression, may be justified by providing a derivation of the sequence of terminals from the non-terminal $\langle \text{Expression} \rangle$. For example, a derivation of 3-4 is:

$$\begin{aligned}\langle \text{Expression} \rangle &\Rightarrow \langle \text{Expression} \rangle - \langle \text{Expression} \rangle \\ &\Rightarrow \langle \text{Number} \rangle - \langle \text{Expression} \rangle \\ &\Rightarrow 3 - \langle \text{Expression} \rangle \\ &\Rightarrow 3 - \langle \text{Number} \rangle \\ &\Rightarrow 3 - 4\end{aligned}$$

Each step of the derivation results from unfolding a production of the grammar given above. Examples of syntactically incorrect expressions are 3--4, 4 div 2, and 3-().

1.1.2 Abstract Syntax

Given a string of terminals s , a parser will produce an abstract syntax tree (AST) for s if it is a syntactically correct ARITH expression, otherwise it will fail with an error message. The AST is a value of type `prog`. A value of type `prog` is an expression of the form `AProg(cs,e)` where `cs` is a list of class declarations and `e` is an expression of type `expr`. As discussed below, `cs` will be empty for now. Indeed, we shall only be interested in `e` for the moment:

```
type
2  prog = AProg of (cdecl list)*expr
and
4  expr =
    | Int of int
    | Sub of expr*expr
    | Div of expr*expr
```

[parser_plaf/lib/ast.ml](#)



This figure is an example of a code listing. It occasionally includes an indication as to where the snippet of code resides. For example, in this case it resides in file [parser_plaf/lib/ast.ml](#).

The `parse` function parses a string. If the string obeys the rules of the concrete syntax, it produces an associated abstract syntax tree; otherwise it reports an error. For example,

```

# parse "1+2*3";;
- : prog = AProg ([], Add (Int 1, Mul (Int 2, Int 3)))

```

utop

Notice that `parse` has type `string->prog` rather than `string->expr`. As mentioned above, a program is an expression of the form `AProg(cs,e)`, where `cs` is a list of class declarations and `e` is an expression of type `expr`. We shall ignore class declarations for now and discuss them in detail in Chapter 5. In the meantime, our programs will always have the form `AProg([],e)` and we shall only be interested in evaluating the expression `e`.

1.2 Interpreter

An interpreter¹ is a process that, given an expression, produces the result of its evaluation. The implementation of interpreters in these notes will be developed in two steps, first we specify the interpreter and then we implement it proper.

- Specification of the interpreter. This consists in first providing a precise description of the possible results which a program can evaluate to. Then introducing evaluation judgements that state what result a program evaluates to. Finally, a set of evaluation rules is introduced that define the meaning of evaluation judgements by describing the behavior of each construct in the language.
- Implementation of the interpreter. Using the evaluation rules of the specification as a guide, an implementation is presented. The time invested in producing the evaluation rules in the previous step, betters our understanding of the interpreter's behavior and hence diminishes the chances of introducing errors when it is implemented.

To illustrate this approach, we next specify an interpreter for ARITH and then implement it.

1.2.1 Specification

We begin by stating the possible results that can arise out of the evaluation of programs in ARITH. For now, we fix the set of **results** to be the integers \mathbb{Z} since evaluation of ARITH programs produce integers:

$$\mathbb{R} := \mathbb{Z}$$

The “ $:=$ ” symbol is used for definitional equality, meaning here that \mathbb{R} is “defined to be” the set \mathbb{Z} of integers. We continue with the specification of the interpreter for ARITH by introducing evaluation judgements. **Evaluation judgements** are expressions of the form:

$$e \Downarrow n$$

where `e` is an expression in ARITH in abstract syntax and `n` is a result (*i.e.* `n` is an integer). Evaluation judgements are read as, “expression `e` evaluates to the integer `n`”. The meaning of `e \Downarrow n` is established via so called **evaluation rules**. The preliminary set of evaluation rules of ARITH are as follows:

¹We will use the words “interpreter” and “evaluator” interchangeably.

$$\begin{array}{c}
\frac{}{\text{Int}(n) \Downarrow n} \text{EInt} \\
\\
\frac{e1 \Downarrow m \quad e2 \Downarrow n \quad p = m - n}{\text{Sub}(e1, e2) \Downarrow p} \text{ESub} \quad \frac{e1 \Downarrow m \quad e2 \Downarrow n \quad p = m/n}{\text{Div}(e1, e2) \Downarrow p} \text{EDiv}
\end{array}$$

Judgements above the horizontal line in an evaluation rule are called **hypotheses** and the one below is called the **conclusion**. A rule that does not have hypotheses is called an **axiom rule** or just **axiom**. Hypotheses of an evaluation rule are read from left to right. In particular, evaluation of the arguments of all arithmetic operations proceeds from left to right. It could have been stated in the opposite order from right-to-left and, at this point in time, does not make much of a difference². A **derivation tree** is a tree of evaluation judgements whose nodes are instances of evaluation rules and, moreover, whose leaves are instances of axioms. A judgement $e \Downarrow n$ is **derivable** or is **said to 'hold'** if there is a derivation tree with $e \Downarrow n$ as its root.

Example 1.2.1. For example $\text{Sub}(\text{Int } 3, \text{Int } 1) \Downarrow 2$ is a derivable evaluation judgement:

$$\frac{\frac{}{\text{Int}(3) \Downarrow 3} \text{EInt} \quad \frac{}{\text{Int}(1) \Downarrow 1} \text{EInt} \quad 2 = 3 - 1}{\text{Sub}(\text{Int } 3, \text{Int } 1) \Downarrow 2} \text{ESub}$$

The evaluation judgement $\text{Sub}(\text{Int } 3, \text{Int } 1) \Downarrow 1$ is not derivable. The evaluation judgement $\text{Sub}(\text{Div}(\text{Int } 8, \text{Int } 2), \text{Int } 1) \Downarrow 3$ is also derivable.

We are not quite done with the task of specifying our interpreter since not all expressions in ARITH return numbers. For example, $\text{Div}(\text{Int } 2, \text{Int } 0)$ does not evaluate to any number. Rather it should evaluate to an error. Thus, the above mentioned evaluation rules are incomplete since there are situations that are left unspecified. It is important to have a complete set of rules so that when we implement the interpreter there are no ambiguities. Moreover, we must revisit our notion of result since it should include an error as a possible outcome. A result is either an integer or a special element *error*:

$$\mathbb{R} := \mathbb{Z} \cup \{\text{error}\}$$

The subset of results that are integers are called expressed values: it is the name given to non-error results of evaluation. The previously introduced evaluation judgements are thus revisited below. The final form that evaluation judgements take for ARITH are:

$$e \Downarrow r$$

where r denotes a result of the computation $r \in \mathbb{R}$.

The evaluation rules defining this new judgement, and therefore the evaluation rules for ARITH, are those presented in Figure 1.1, where $m, n, p \in \mathbb{Z}$. The first three rules are the ones already presented above. Rules ESubErr1, ESubErr2, EDivErr1, and EDivErr2 state how error propagation takes place. The last one introduces a new error, namely division by zero. Moving forward, and for the sake of brevity, we will not be specifying the error propagation rules when specifying the interpreter. We will only be presenting the error introduction rules.

²But later in our development, when evaluation of expressions can cause certain effects (such as modifying mutable data structures), this difference will become relevant.

$$\begin{array}{c}
\frac{}{\text{Int}(n) \Downarrow n} \text{EInt} \\
\\
\frac{e1 \Downarrow m \quad e2 \Downarrow n \quad p = m - n}{\text{Sub}(e1, e2) \Downarrow p} \text{ESub} \quad \frac{e1 \Downarrow m \quad e2 \Downarrow n \quad n \neq 0 \quad p = m/n}{\text{Div}(e1, e2) \Downarrow p} \text{EDiv} \\
\\
\frac{e1 \Downarrow \text{error}}{\text{Sub}(e1, e2) \Downarrow \text{error}} \text{ESubErr1} \quad \frac{e1 \Downarrow m \quad e2 \Downarrow \text{error}}{\text{Sub}(e1, e2) \Downarrow \text{error}} \text{ESubErr2} \\
\\
\frac{e1 \Downarrow \text{error}}{\text{Div}(e1, e2) \Downarrow \text{error}} \text{EDivErr1} \quad \frac{e1 \Downarrow m \quad e2 \Downarrow \text{error}}{\text{Div}(e1, e2) \Downarrow \text{error}} \text{EDivErr2} \quad \frac{e1 \Downarrow m \quad e2 \Downarrow 0}{\text{Div}(e1, e2) \Downarrow \text{error}} \text{EDivErr3}
\end{array}$$

Figure 1.1: Evaluation rules for ARITH

Example 1.2.2. The evaluation judgement $\text{Sub}(\text{Div}(\text{Int } 8, \text{Int } 0), \text{Int } 1) \Downarrow \text{error}$ is derivable. Just like in the previous example, the evaluation judgement $\text{Sub}(\text{Int } 3, \text{Int } 1) \Downarrow 1$ is not derivable.

We next address the implementation of an interpreter for ARITH. We will do so in two attempts, first a preliminary attempt and then a final one. The preliminary attempt has various drawbacks that we will point out along the way but has the virtue of serving as a convenient stepping stone towards the final one.



A summary of some important concepts we have covered are listed below. Make sure you look them up above:

- Result
- Expressed Value
- Evaluation Judgement
- Evaluation Rules
- Derivation Tree
- Derivable Evaluation Judgements

1.2.2 Implementation

In order to use the evaluation rules as a guideline for our implementation we first need to model both components of evaluation judgements in OCaml, namely expression e and result r in $e \Downarrow r$. The former is already expressed in abstract syntax, which we encoded as the algebraic data type `expr` in OCaml. So that item has already been addressed. As for the latter, since it denotes

```

let rec eval_expr : expr -> int result =
2  fun e ->
    match e with
4  | Int(n) -> Ok n
    | Sub(e1,e2) ->
6      (match eval_expr e1 with
         | Error s -> Error s
         | Ok m -> (match eval_expr e2 with
8             | Error s -> Error s
             | Ok n -> Ok (m-n)))
    | Div(e1,e2) ->
12     (match eval_expr e1 with
        | Error s -> Error s
        | Ok m -> (match eval_expr e2 with
14             | Error s -> Error s
             | Ok n -> if n==0
16                 then Error "Division by zero"
                 else Ok (m/n)))
18

```

interp.ml

Figure 1.2: Preliminary Interpreter for ARITH

either an integer or an error, we will model it in OCaml using the following type³:

```

type 'a result = Ok of 'a | Error of string

```

ds.ml

For example, the type `int result` may be read as a type that states that “the result of the evaluator is an integer or an error”. Likewise, `bool result` may be read as a type that states that “the result of the evaluator is a boolean or an error”. In summary, a result can either be a meaningful value of type `'a` prefixed with the constructor `Ok`, or else an error with an argument of type `string` prefixed with an `Error` constructor. For example, `Ok 3` has type `int result`.



In a type expression such as `int result`, we say `int` is a “type” and `int result` is a “type”. But we refer to `result` as a **type constructor** since, given a type `'a`, it constructs a type `'a result`.

We can now proceed with an implementation of an interpreter for ARITH following the evaluation rules as close as possible. If we call our evaluator function `eval_expr`, its type can be expressed as follows, indicating that evaluation consumes an expression and returns either an integer or an error with a string description:

```
eval_expr : exp -> int result
```

The code is given in Figure 1.2. The `eval_expr` function is defined by recursion over the structure of expressions in abstract syntax (*i.e.* values of type `expr`). In the first clause, `Int(n)`, it simply returns `Ok n`. Note that returning just `n` would be incorrect since our interpreter must return a value of type `int result`, not of type `int`. In the clause for `Sub(e1,e2)`, the `match` keyword

³OCaml has a built-in type `type ('a,'b) result = Ok of 'a | Error of 'b`. We could have used this type but it is slightly more general than necessary since our errors will always be accompanied by a string argument rather than different types of arguments.

forces evaluation of e_1 before e_2 , as indicated by the evaluation rules⁴. A similar comment applies to the other arithmetic operation. One notices that a substantial amount of code checks for errors and then propagates them. Notice too that, in the `Div` case, in addition to propagating errors resulting from evaluating its arguments e_1 and e_2 , it generates a new one if the denominator is 0. This is in accordance with the evaluation rule `EDivErr3`. The only error modeled in `ARITH` is division by zero. An example of error propagation takes place in an `ARITH` expression such as `Add(Div(Int 1, Int 0), e)`, where e can be any expression. Here the expression e is never evaluated since evaluation of `Div(Int 1, Int 0)` produces `Error "Division by zero"`, hence e is ignored and `Error "Division by zero"` is immediately produced as the final result of evaluation of the entire expression.

1.2.2.1 Implementation: Final

Although certainly necessary, there is no interesting computational content in error propagation. It would be best to have it be handled behind the scenes, by appropriate error propagation helper functions. We next introduce three helper functions for this purpose:

- `return`,
- `error`, and
- `(>>=)` (pronounced “bind”).

The code for these functions is given in Figure 1.3. The `return` function simply returns its argument inside an `Ok` constructor and may thus be seen as producing a non-error result. A similar comment applies to `error`: given a string it produces an error result by simply prefixing the string with the `Error` constructor. The infix operator `(>>=)` is called `bind` and is left associative⁵. Consider the expression `c >>= f`; its behavior may be described as follows:

1. evaluate the argument c to produce a result (*i.e.* a non-error value or an error value); if c returns an error, propagate it and conclude.
2. otherwise, if c returns `Ok v`, for some expressed value v , then pass v on to f by evaluating `f v`.

An alternative description of these helper functions is as follows. Let us dub expressions of type `int result`, **structured programs** (we could have taken the more general type `'a result` as our notion of structured programs, but the latter will suffice for our explanation). Structured programs may be seen as programs that, apart from producing an integer as end product, can manipulate additional structure such as error handling, state, non-determinism, etc. In our particular case, a structured program handles errors as additional structure. Under this light, we can describe the helper functions as follows:

- `return` may be seen as a function that creates a (trivial) structured program that returns an integer (*i.e.* non-error) result.

⁴OCaml evaluates arguments from right to left.

⁵The precedence and associativity of user-defined infix/prefix operators may be consulted here: <https://caml.inria.fr/pub/docs/manual-caml-light/node4.9.html>

```

let return : 'a -> 'a result =
2   fun v -> Ok v

let error : string -> 'a result =
4   fun s -> Error s

let (>=) : 'a result -> ('a -> 'b result) -> 'b result =
8   fun c f ->
    match c with
10  | Error s -> Error s
    | Ok v -> f v

```

ds.ml

Figure 1.3: The Error Monad

- error may be seen as a function that creates a (trivial) structured program that returns an error result.
- (>=) may be seen as a means of composing structured programs. In `c >= f`, structured program `c` is composed with structured program `f v`, where `v` is the non-error result of `c`. If `c` produces an error, then evaluation of `f` is skipped.

Let us rewrite our interpreter for our simple expression language using these helper functions.

```

let rec eval_expr : expr -> int result =
2   fun e ->
    match e with
4   | Int(n) -> return n
    | Sub(e1,e2) ->
6       eval_expr e1 >= (fun n1 ->
7         eval_expr e2 >= (fun n2 ->
8           return (n1-n2)))
    | Div(e1,e2) ->
10      eval_expr e1 >= (fun n1 ->
11        eval_expr e2 >= (fun n2 ->
12          if n2==0
13            then error "Division by zero"
14            else return (n1/n2)))

```

interp.ml

Consider the code for the `Sub(e1,e2)` case. Notice how if `eval_expr e1` produces an error result, say `Error "Division by zero"` because `e1` had a division by zero, then `(>=)` simply ignores its second argument, namely the expression `(fun n1 -> eval_expr e2 >= (fun n2 -> return (n1+n2)))`, and returns the error result `Error "Division by zero"` immediately as the final result of the evaluation, thus effectively propagating the error.

In fact, we can further simplify this code by dropping superfluous parenthesis. This leads to our final evaluator for ARITH expressions.

```

let rec eval_expr : expr -> int result =
2   fun e ->
    match e with
4   | Int(n) -> return n
    | Sub(e1,e2) ->
6       eval_expr e1 >= fun n1 ->

```



```

8   eval_expr e2 >>= fun n2 ->
    return (n1-n2)
|   Div(e1,e2) ->
10  eval_expr e1 >>= fun n1 ->
    eval_expr e2 >>= fun n2 ->
12  if n2==0
    then error "Division by zero"
14  else return (n1/n2)

```

Some additional observations on the behavior of the error handling operations:

$$\begin{aligned}
 \text{return } e \gg= f &\simeq f\ e \\
 m \gg= \text{return} &\simeq m \\
 (m \gg= f) \gg= g &\simeq m \gg= (\text{fun } x \rightarrow f\ x \gg= g) \\
 \text{error } e \gg= f &\simeq \text{error } e
 \end{aligned}$$

The symbol \simeq above means that the left and right hand sides of these equations behave the same way.



The `result` type, together with the operations `return`, `error` and `(>>=)` is called an **Error Monad**. Monads are well-known in pure functional programming languages like Haskell, where they allow to handle side-effects behind the scenes without compromising equational reasoning (see the equations presented above). However, they are also important in non-pure functional languages, like OCaml, where they are a means to better structure one's code, as we have seen from our use of it here.

1.3 Exercises

Exercise 1.3.1. For each of the following sequence of terminals, write a derivation that shows that it belongs to the grammar generated by the nonterminal `<Expression>`:

1. `(3-4)`
2. `((4/0)-4)`
3. `3-4-1`

Exercise 1.3.2. What is the difference between a result and an expressed value?

Exercise 1.3.3. Consider the extension of ARITH with a new expression that returns the absolute value of the value of its argument. The concrete syntax of ARITH is extended with the production:

$$\langle \text{Expression} \rangle ::= \text{abs}(\langle \text{Expression} \rangle)$$

and the abstract syntax with a new constructor:

```

type expr =
| Int of int
| Sub of expr*expr
| Div of expr*expr
| Abs of expr

```

`parser_plaf/lib/ast.ml`

1. *Do the set of results in this extended language need to be modified? Think about whether new errors are introduced by the `abs` construct or whether new kinds of expressed values (other than integers) are possible.*
2. *Do evaluation judgements need to be modified? Think about whether evaluation judgements still have the form $e \Downarrow r$, for e an expression in the extended language and r a result.*
3. *Add the two evaluation rules for the new language construct `abs(e)`. You may assume that `abs` is the name of the mathematical function that returns the absolute value of an integer.*
4. *Extend the interpreter `eval_expr` to handle this case.*

Chapter 2

Simple Functional Languages

2.1 LET

2.1.1 Concrete Syntax

```
⟨Expression⟩ ::= ⟨Number⟩
⟨Expression⟩ ::= ⟨Identifier⟩
⟨Expression⟩ ::= ⟨Expression⟩⟨BOp⟩⟨Expression⟩
⟨Expression⟩ ::= zero?(⟨Expression⟩)
⟨Expression⟩ ::= if ⟨Expression⟩ then ⟨Expression⟩ else ⟨Expression⟩
⟨Expression⟩ ::= let ⟨Identifier⟩ = ⟨Expression⟩ in ⟨Expression⟩
⟨Expression⟩ ::= (⟨Expression⟩)

⟨BOp⟩          ::= + | - | * | /
```

Note that we have not specified what terminals are generated by the `⟨Identifier⟩` non-terminal; we will assume these to be sequences of symbols 'a'-'z', 'A'-'Z', '0'-'9', or '_' that start with a letter.

2.1.2 Abstract Syntax

```
1 type expr =
  | Int of int
  | Var of string
  | Add of expr*expr
  | Sub of expr*expr
  | Mul of expr*expr
  | Div of expr*expr
  | IsZero of expr
  | ITE of expr*expr*expr
  | Let of string*expr*expr
```

2.1.3 Environments

Consider the LET expression $x+2$. Variables such as x are referred to as identifiers. We cannot determine the result of evaluating this expression because, in a sense, it is incomplete. Indeed, unless we are given the value assigned to the identifier x , we cannot determine whether evaluation of $x+2$ should result in an error (if say, x held the value true ¹), or a number such as 4 (if, say, x held the value 2). Therefore, evaluation of expressions in LET require an assignment of values to identifiers. These assignments are called environments. The interpreters developed in these notes are therefore known as environment-based interpreters as opposed to substitution-based interpreters. In the latter values are substituted directly into the expressions rather than recording, and then looking them up, in environments.

An environment is a partial function from the set of identifiers to the set of expressed values. Expressed values, denoted \mathbb{EV} , are the set of values that are not errors that we can get from evaluating expressions. In ARITH the only expressed values are the integers. In LET they are the integers and the booleans:

$$\mathbb{EV} := \mathbb{Z} \cup \mathbb{B}$$

where $\mathbb{B} := \{\text{true}, \text{false}\}$. If \mathbb{ID} denotes the set of all identifiers², then we can define the set of all environments \mathbb{ENV} as follows.

$$\mathbb{ENV} := \mathbb{ID} \rightarrow \mathbb{EV}$$

We use letters ρ and ρ' to denote environments. For example, $\rho = \{x := 1, y := 2, z := \text{true}\}$ is an environment that assigns 1 to x , 2 to y and true to z . We write $\rho(id)$ for the value associated to the identifier id . For example, $\rho(x)$ is 1.

2.1.4 Interpreter

2.1.4.1 Specification

As you might recall from our presentation of ARITH, evaluation of an ARITH expression produces a result. A result could either be an integer or an error. We can still get an error from evaluating a LET expression since ARITH expressions are included in LET expressions. However, if there is no error, then in LET we can either get an integer or a boolean. The set of results for LET is thus:

$$\mathbb{R} := \mathbb{EV} \cup \{\text{error}\}$$

where \mathbb{EV} was updated above during our discussion on environments. Evaluation judgements for LET include an environment:

$$e, \rho \Downarrow r$$

It should be read as follows, “evaluation of expression e under environment ρ , produces result r ”. The rules defining this judgement are presented in Figure 2.1. The last four rules handle error generation, the first eight handle standard (i.e. non-error) evaluation. The rules for error propagation are omitted. Rule EInt is the same as in ARITH, except that the judgement has

¹We do not have booleans in ARITH but we will in LET.

²The elements of \mathbb{ID} are assumed to be the same as those generated by the non-terminal $\langle \text{Identifier} \rangle$.

$$\begin{array}{c}
\frac{}{\text{Int}(n), \rho \Downarrow n} \text{EInt} \quad \frac{\rho(\text{id}) = v}{\text{Var}(\text{id}), \rho \Downarrow v} \text{EVar} \\
\\
\frac{\text{e1}, \rho \Downarrow m \quad \text{e2}, \rho \Downarrow n \quad n \neq 0 \quad p = m/n}{\text{Div}(\text{e1}, \text{e2}), \rho \Downarrow p} \text{EDiv} \\
\\
\frac{\text{e}, \rho \Downarrow 0}{\text{IsZero}(\text{e}), \rho \Downarrow \text{true}} \text{EIZTrue} \quad \frac{\text{e}, \rho \Downarrow m \quad m \neq 0}{\text{IsZero}(\text{e}), \rho \Downarrow \text{false}} \text{EIZFalse} \\
\\
\frac{\text{e1}, \rho \Downarrow \text{true} \quad \text{e2}, \rho \Downarrow v}{\text{ITE}(\text{e1}, \text{e2}, \text{e3}), \rho \Downarrow v} \text{EITETTrue} \quad \frac{\text{e1}, \rho \Downarrow \text{false} \quad \text{e3}, \rho \Downarrow v}{\text{ITE}(\text{e1}, \text{e2}, \text{e3}), \rho \Downarrow v} \text{EITEFalse} \\
\\
\frac{\text{e1}, \rho \Downarrow w \quad \text{e2}, \rho \oplus \{\text{id} := w\} \Downarrow v}{\text{Let}(\text{id}, \text{e1}, \text{e2}), \rho \Downarrow v} \text{ELet} \\
\\
\frac{\text{id} \notin \text{dom}(\rho)}{\text{Var}(\text{id}), \rho \Downarrow \text{error}} \text{EVarErr} \quad \frac{\text{e1}, \rho \Downarrow m \quad \text{e2}, \rho \Downarrow 0}{\text{Div}(\text{e1}, \text{e2}), \rho \Downarrow \text{error}} \text{EDivErr} \\
\\
\frac{\text{e}, \rho \Downarrow v \quad v \notin \mathbb{Z}}{\text{IsZero}(\text{e}), \rho \Downarrow \text{error}} \text{EIZErr} \quad \frac{\text{e1}, \rho \Downarrow v \quad v \notin \mathbb{B}}{\text{ITE}(\text{e1}, \text{e2}, \text{e3}), \rho \Downarrow \text{error}} \text{EITEErr}
\end{array}$$

Figure 2.1: Evaluation Semantics for LET (error propagation rules omitted)

an environment (which plays no role in this rule). Rule EVar performs lookup in the current environment. The related rule EVarErr models the error resulting from lookup failing to find the identifier in the environment. The rules for addition, subtraction and multiplication are similar as the one for division and omitted. The notation $\rho \oplus \{\text{id} := w\}$ used in ELet stands for the environment that maps expressed value w to identifier id and behaves as ρ on all other identifiers.

2.1.4.2 Implementation: Attempt I

Before implementing the evaluator we must first implement expressed values and environments. Expressed values can be naturally described using algebraic data types. Environments can be modeled in various ways in OCaml: as functions, as association lists, as hash tables, as algebraic data types, etc. Due to its simplicity we follow the last of these.

```

type exp_val =
2   | NumVal of int
   | BoolVal of bool
4
type env =
6   | EmptyEnv
   | ExtendEnv of string*exp_val*env

```

ds.ml

Operations on environments are:

```

let rec eval_expr : expr -> env -> exp_val result =
2  fun e en ->
    match e with
4  | Int(n) -> return (NumVal n)
    | Var(id) -> apply_env id en
6  | Div(e1,e2) -> (* Add, Sub and Mul are similar and omitted *)
        eval_expr e1 en >=>
8      int_of_numVal >=> fun n1 ->
        eval_expr e2 en >=>
10     int_of_numVal >=> fun n2 ->
        if n2==0
12     then error "Division by zero"
        else return (NumVal (n1/n2))
14 | IsZero(e) ->
        eval_expr e en >=>
16     int_of_numVal >=> fun n ->
        return (BoolVal (n = 0))
18 | ITE(e1,e2,e3) ->
        eval_expr e1 en >=>
20     bool_of_boolVal >=> fun b ->
        if b
22     then eval_expr e2 en
        else eval_expr e3 en
24 | Let(id,def,body) ->
        eval_expr def en >=> fun ev ->
26     eval_expr body (extend_env en id ev)
    | _ -> failwith "Not implemented yet!"

```

interp.ml

Figure 2.2: Preliminary Interpreter for LET

```

1  let empty_env : unit -> env =
    fun () -> EmptyEnv
3
let extend_env : env -> string -> exp_val -> env =
5  fun env id v -> ExtendEnv(id,v,env)

7  let rec apply_env : string -> env -> exp_val result =
    fun id env ->
9      match env with
    | EmptyEnv -> error (id^" not found!")
11   | ExtendEnv(v,ev,tail) ->
        if id=v
13     then return ev
        else apply_env id tail

```

ds.ml

Notice that `apply_env en id` has `exp_val result` as return type because it returns an error if `id` is not found in the environment `en`. However, if there is an expressed value associated to `id` in `en`, then that will be returned (wrapped with an `OK` constructor).

Next we implement an interpreter for LET by following the evaluation rules of Figure 2.1 closely. Indeed, the evaluation rules shall serve as a specification for, and thus guide, our implementation. The code itself is given in Figure 2.2. The case for `Var(id)` simply invokes `apply_env` to look up the expressed value associated to the identifier `id` in the environment `env`.

The case for `Div(e1,e2)` makes use of an auxiliary operation `int_of_numVal`, explained below.

```

2 let int_of_numVal : exp_val -> int result =
  fun ev ->
4   match ev with
    | NumVal n -> return n
    | _ -> error "Expected a number!"

```

[ds.ml](#)

Evaluation of `e1` in `Div(e1,e2)` could produce an expressed value other than a number (*i.e.* other than a `NumVal`). The function `int_of_numVal` checks to see whether its argument is a `NumVal` or not, returning a result that consists of an error, if it is not, or else the number itself (without the `NumVal` tag). This number is then bound to the variable `n1`. A similar description determines the value of `n2`. Finally, if `n2` is zero, an error is returned, otherwise the desired quotient is produced as a result.

The code for the other binary operators, for the zero predicate and for the conditional are similar. The case for `ITE` uses a similar helper function `bool_of_boolVal`. The last case, `Let`, evaluates the definition and then extends the environment appropriately before evaluating the body. Notice how local scoping is implemented by adding a new entry into the environment.



The default case, in the last line of Figure 2.2, handles language constructs that the parser supports but will only be explained later. This case simply reports that they are not implemented yet. Notice how the OCaml runtime `Failure` exception is used to report an error, rather than the `error` operation reserved for object-level (*i.e.* those arising from errors resulting from the execution of the `LET`) errors.

The top level function for the interpreter is called `interp`. It parses the string argument, evaluates producing a function `c` that awaits an environment, and then feeds that function the empty environment `EmptyEnv`.

```

2 let interp (s:string) : exp_val result =
  let c = s |> parse |> eval_expr
  in c EmptyEnv

```

[interp.ml](#)

Here is an example run of our interpreter³:

```

# interp "
2 let x=2
  in let y=3
4   in x+y";;
- : exp_val Ds.result = Ok (NumVal 5)

6
utop # interp "
8 let x=2
  in let y=0
10  in x+(x/y)";;
- : exp_val Ds.result = Error "Division by zero"

```

[utop](#)

³Negative numbers must be placed between parenthesis. For example, `interp "(-7)"` rather than `interp "-7"`.



In ARITH there was only one possible error that could be generated and then propagated, namely the division by zero error. In LET there are four possible errors that can be generated and propagated: division by zero, identifier not found, expected a number and expected a boolean.

2.1.4.3 Weaving Environments

Our code for LET seems well-structured and robust enough to be extensible to additional language features. Even so, we can perhaps take a step further. Notice that the environment is explicitly threaded around the entire program. Indeed, consider the following excerpt from Figure 2.2 and notice how the environment (highlighted) is passed on to each occurrence of `eval_expr`.

```

let rec eval_expr : expr -> env -> exp_val result =
2   fun e en ->
    match e with
4   | ...
    | ITE(e1,e2,e3) ->
6     eval_expr e1 en >>=
      bool_of_boolVal >>= fun b ->
8     if b
      then eval_expr e2 en
10    else eval_expr e3 en

```

The reason `en` is passed on in each case above, is that all expressions `e1`, `e2` and `e3` are evaluated under that same environment. This occurs with other language constructs too such as `Add(e1,e2)`, `Div(e1,e2)`, `Sub(e1,e2)` and `Mul(e1,e2)`, where both `e1` and `e2` are evaluated under the environment `en`. An alternative would be to have the environment be passed around “behind the scenes”, in the same way that error propagation is handled behind the scenes. The resulting code would look something like this, where all references to the environment have been removed, including the one on line 2:

```

let rec eval_expr : expr -> env -> exp_val result =
2   fun e ->
    match e with
4   | ...
    | ITE(e1,e2,e3) ->
6     eval_expr e1 >>=
      bool_of_boolVal >>= fun b ->
8     if b
      then eval_expr e2
10    else eval_expr e3

```

Listing 2.3: Naive removal of environment arguments

We would still need to provide an environment since `eval_expr` expects both expression and environment. That would be done by `interp`:

```

let interp (s:string) : exp_val result =
2   let c = s |> parse |> eval_expr
    in c EmptyEnv

```

Listing 2.4: Naive removal of environment arguments

Unfortunately, the resulting code in Listing 2.3 doesn't type-check. Let us take a closer look at the bind operator used in line 6:

(2.1)

`eval_expr e1 >>= ...`

Recall from Figure 1.3 that the type of `(>>=)` is

```
(>>=) : 'a result -> ('a -> 'b result) -> 'b result
```

The expression `eval_expr e1` in (2.1) is therefore expected to have type `'a result` (where `'a` can be any type, in particular `exp_val`). However, since we removed the environment argument it instead has type `env -> exp_val result`. Indeed, `eval_expr e1` now produces a:

function that waits for the environment and then produces a result.

This means that `bind` now has to be able to compose “functions that wait for environments and produce a result” rather than composing “results”. In other words, we have to put forward a new proposal for the type of `bind`:

```
Currently      (>>=) : 'a result -> ('a -> 'b result) -> 'b result
New proposal   (>>=) : (env -> 'a result) -> ('a -> (env -> 'b result)) -> (env -> 'b result).
```

Lets give the type `env -> 'a result` a name, so that we can improve legibility of the type expression above. Consider the following new `ea_result` type constructor, read “environment abstracted result”, defined by simply abstracting the type of environments over the standard result type:

```
type 'a ea_result = env -> 'a result
```

Now we can apply this type synonym and recast our table above as:

```
Currently      (>>=) : 'a result -> ('a -> 'b result) -> 'b result
New proposal   (>>=) : 'a ea_result -> ('a -> 'b ea_result) -> 'b ea_result.
```

Of course, we’ll need to update the code for `bind` (and the other helper functions). We will do so shortly. Applying the type synonym again, this time to the type of `eval_expr`, the new type for our interpreter is now:

```
Currently      eval_expr : expr -> env -> exp_val result
New proposal   eval_expr : expr -> exp_val ea_result
```

Updating the helper functions. Since the type for the helper functions such as `bind` has changed, we must now update their code. The new code for them is in Figure 2.5. Function `return v` used to return `Ok v`. But notice now how it returns a function that waits for an environment `env` and only then returns `Ok v`. It may perhaps result odd that the environment seems not to be used for anything. However, other helper functions will make use of it (for example, `(>>=)`). Note also how `(>>=)` now passes the environment argument `env` first to `c` and then to `f v`, thus effectively threading the environment for us. You may safely ignore `(>>+)` for now, we’ll explain it later. Also, we have a new operation `run` that given an environment abstracted result, will feed it the empty environment and thus perform the computation itself resulting in either an `ok` value or an error value. It is essentially the same as Listing 2.4 except that, since this function will be placed in the file `interp.ml`, it is best to avoid using the names of the constructors for environments.

```
2 type 'a result = Ok of 'a | Error of string
4
6 type 'a ea_result = env -> 'a result
8
10 let return : 'a -> 'a ea_result =
12   fun v ->
14     fun env -> Ok v
16
18 let error : string -> 'a ea_result =
20   fun s ->
22     fun env -> Error s
24
26 let (>=) : 'a ea_result -> ('a -> 'b ea_result) -> 'b ea_result =
28   fun c f ->
     fun env ->
       match c env with
       | Error err -> Error err
       | Ok v -> f v env
20
22 let (>+) : env ea_result -> 'a ea_result -> 'a ea_result =
24   fun c d ->
26     fun env ->
28       match c env with
       | Error err -> Error err
       | Ok newenv -> d newenv
20
22 let run : 'a ea_result -> 'a result =
24   fun c -> c EmptyEnv
```

ds.ml

Figure 2.5: The Reader and Error Monad Combined

```

2 let interp (e:string) : exp_val result =
  let c = e |> parse |> eval_expr
  in run c

```

[interp.ml](#)

The variable `c` is used as mnemonic for “computation” (also referred to as a “structured program”) the program that results from evaluating the abstract syntax tree of `e`. The computation is executed by passing it the empty environment.

2.1.4.4 Implementation: Final

We next revisit our evaluator for LET, this time making use of our new environment abstracted result type. The code is given in Figure 2.6. We briefly comment on some of the variants.

The code for the `Int(n)` variant, remains unaltered:

```
| Int(n) -> return (NumVal n)
```

Note, however, that `return (NumVal n)` now returns a function that given an environment, ignores it and simply returns `Ok (NumVal n)`.

The `Var(id)` variant is similar, it is missing the environment:

```
| Var(id) -> apply_env id
```

Now `apply_env` is applied only to the argument `id`, thus producing an expression (through partial application) that waits for the second argument, namely the environment. This environment will be supplied when we run the computation (using `run`).

The variants `Div(e1,e2)`, `IsZero(e)` and `ITE(e1,e2,e3)` are as in Figure 2.2 except that the environment argument has been dropped. Finally, consider `Let(id,def,body)`. Let us recall from Figure 2.2, the code we had for this variant:

```

2 | Let(id,def,body) ->
  eval_expr def en >>= fun ev ->
  eval_expr body (extend_env en id ev)

```

We first evaluate `def` in the current environment `en` producing an expressed value `ev`. This expressed value is used to extend the current environment `ev`, before evaluating the body `body`. Dropping the environment arguments, which are now threaded implicitly for us, results in:

```

2 | Let(id,def,body) ->
  eval_expr def en >>= fun ev ->
  eval_expr body (extend_env id ev)

```

There are two problems with this code. First we need to be able to produce the modified environment resulting from adding the new key value-pair `id:=ev` into environment `en` **as a result** so that we can pass it on when evaluating `body`. This is achieved by updating `extend_env`, and `empty_env` too although we will not be needing the latter for now, that produces the updated environment as a result (notice the `env` in `env ea_result`):

```

2 let extend_env : string -> exp_val -> env ea_result =
  fun id v ->
  fun env -> Ok (ExtendEnv(id,v,env))

```

```

let rec eval_expr : expr -> exp_val ea_result =
2   fun e ->
    match e with
4   | Int(n) -> return (NumVal n)
    | Var(id) -> apply_env id
6   | Div(e1,e2) -> (* Add, Sub and Mul are similar and omitted *)
        eval_expr e1 >>=
8       int_of_numVal >>= fun n1 ->
        eval_expr e2 >>=
10      int_of_numVal >>= fun n2 ->
        if n2==0
12      then error "Division by zero"
        else return (NumVal (n1/n2))
14  | IsZero(e) ->
        eval_expr e >>=
16      int_of_numVal >>= fun n ->
        return (BoolVal (n = 0))
18  | ITE(e1,e2,e3) ->
        eval_expr e1 >>=
20      bool_of_boolVal >>= fun b ->
        if b
22      then eval_expr e2
        else eval_expr e3
24  | Let(id,def,body) ->
        eval_expr def >>=
26      extend_env id >>+
        eval_expr body
28  | _ -> failwith "Not implemented yet!"

30 let parse s =
    let lexbuf = Lexing.from_string s in
32    let ast = Parser.prog Lexer.read lexbuf in
    ast
34
let interp (e:string) : exp_val result =
36  let c = e |> parse |> eval_expr
  in run c

```

interp.ml

Figure 2.6: Evaluator for LET

With this new operation we can produce the following code which is almost correct; we still have to discuss what to put in place of `>>???`:

```

1 | Let(id,def,body) ->
2   eval_expr def >>= fun ev ->
3   extend_env id ev >>???
4   eval_expr body

```

which can be simplified to

```

1 | Let(id,def,body) ->
2   eval_expr def >>=
3   extend_env id >>???
4   eval_expr body

```

This code evaluates `def` under the current environment threaded by `bind`, then feeds the resulting expressed value into `extend_env id` to produce an extended environment. But now we are faced with a second problem. It is this extended environment that should be fed into `eval_expr body` and **not** the current environment that is threaded by `bind` (the current environment presumably has no mapping for `id`). This suggests introducing the following environment update operation:

```

1 let (>>+) : env ea_result -> 'a ea_result -> 'a ea_result =
2   fun c d ->
3   fun env ->
4   match c env with
5   | Error err -> Error err
6   | Ok newenv -> d newenv

```

An expression such as `c >>+ d` first evaluates `c env`, where `env` is the current environment, producing an environment `newenv` as a result, and then completely ignores the current environment `env` feeding that new environment as current environment for `d`.

With the help of environment update, we can now complete our code for `Let`:

```

1 | Let(id,def,body) ->
2   eval_expr def >>=
3   extend_env id >>+
4   eval_expr body

```



You can think of `c1 (>>=) f` as a form of composition of computations, “given an environment `en`, pass it on to `c1` producing an expressed value `v`, then pass `v` and `en` on to `f`, and return its result as the overall result”. While `c1 (>>+) c2` may be thought of as, “given an environment `en`, pass it on to `c1` producing an environment `newenv` (not an expressed value!) as a result which is passed on to computation `c2`, returning the latter’s result as the result of the overall computation.”

Note the absence of all references to `en` in Listing. 2.6. Indeed, the environment will be passed around when we execute `run c`. According to the definition of `run`, `run c` just applies `c` to the empty environment `EmptyEnv`.

Example 2.1.1. We conclude this section with some examples of expressions whose type involve the `ea_result` type constructor:

Expression	Type	Informal Description
<code>return (NumVal 3)</code>	<code>exp_val ea_result</code>	Denotes a function that when given an environment, ignores it, and immediately returns <code>Ok (NumVal 3)</code> .
<code>error "oops"</code>	<code>'a ea_result</code>	Denotes a function that when given an environment, ignores it, and immediately returns <code>Error "oops"</code> .
<code>apply_env "x"</code>	<code>exp_val ea_result</code>	Denotes a function that when given an environment, inspects it to find the expressed value v associated to <code>"x"</code> . If it finds it, it returns <code>Ok v</code> , otherwise it returns <code>Error "x not found"</code> .
<code>extend_env "x" (NumVal 3)</code>	<code>env ea_result</code>	Denotes a function that when given an environment, extends it producing a new environment <code>newenv</code> , with the new key-value pair <code>x:=NumVal 3</code> , and returns <code>Ok newenv</code> .

2.1.5 Inspecting the Environment

It is often convenient to be able to inspect the contents of the environment as a means of understanding how evaluation works or simply for debugging purposes. This section extends LET with a new expression `debug(e)` whose evaluation will print the contents of the current environment, ignoring `e` and halting evaluation with an error message. We first add a new production to the grammar defining the concrete syntax of LET:

$$\langle \text{Expression} \rangle ::= \text{debug}(\langle \text{Expression} \rangle)$$

We next add a new variant to the type `expr` defining the abstract syntax of LET:

```
2 type expr =
  ...
  | Debug of expr
```

The next step is to specify, and then implement, the extension to the interpreter for LET that handles the new construct. What should we choose as the value resulting from evaluating `Debug(e)`? In other words, what should we choose to replace the questions marks below with?

$$\frac{}{\text{Debug}(e), \rho \Downarrow ???} \text{EDebug}$$

Since `Debug(e)` has to halt execution (and print the environment), we will have it return an error. This way, no matter where it is placed, the error will get propagated hence effectively halting all further execution. The evaluation rule `EDebug` becomes:

$$\frac{}{\text{Debug}(e), \rho \Downarrow \text{error}} \text{EDebug}$$

Finally, the implementation of this evaluation rule is given below. It makes use of an auxiliary function `string_of_env`, defined in `ds.ml`, which traverses an environment and returns a string representation of it.

```

eval_expr : expr -> exp_val ea_result =
2   fun e ->
    match e with
4   ...
    | Debug(e) ->
6       string_of_env >>= fun str ->
            print_endline str;
8       error "Debug called"

```

Note that there is a slight discrepancy between the specification of the evaluation rule describing how `Debug(e)` is evaluated (*i.e.* `EDebug`) and our implementation. Indeed, the latter prints two strings on the screen but the former does not mention any side-effects such as printing. The reason for this mismatch is that we have decided to keep the specification of our interpreters as simple as possible. In particular, we have decided not to model side-effects such as printing on the screen. Later we will show how to model other side-effecting operations when specifying interpreters. Notably, we will include an assignment operation in our language.

2.2 Exercises

Exercise 2.2.1 (\diamond). Write an OCaml expression of each of the types below:

1. *expr*
2. *env*
3. *exp_val*
4. *exp_val result*
5. *int result*
6. *env result*
7. *int ea_result*
8. *exp_val ea_result*
9. *env ea_result*

Exercise 2.2.2. Consider the following code:

```

open Ds

let c =
  empty_env () >>+
  extend_env "x" (NumVal 1) >>+
  extend_env "y" (BoolVal false) >>+
  string_of_env

```

where the helper function `string_of_env` is defined as follows:

```

let string_of_expval = function
| NumVal n -> "NumVal " ^ string_of_int n
| BoolVal b -> "BoolVal " ^ string_of_bool b

let rec string_of_env' ac = function
| EmptyEnv -> ac
| ExtendEnv(id,v,env) -> string_of_env' ((id^":="^string_of_expval v)::ac) env

let string_of_env : string ea_result =
fun env ->
match env with
| EmptyEnv -> Ok ">>Environment:\nEmpty"
| _ -> Ok (">>Environment:\n" ^ String.concat ",\n" (string_of_env' [] env))

```

ds.ml

1. Knowing that $(>>+)$ associates to the left, fill in all the implicit parenthesis in the definition of c .
2. What is the type of c ?
3. What happens when you load the code into utop and type c ?
4. What happens when you load it into utop and type $run\ c$?

Exercise 2.2.3. Consider the following extension of LET with pairs. Its concrete syntax includes all the grammar productions of LET plus:

$$\begin{aligned}
 \langle \text{Expression} \rangle &::= \text{pair}(\langle \text{Expression} \rangle, \langle \text{Expression} \rangle) \\
 &\quad | \text{fst}(\langle \text{Expression} \rangle) \\
 &\quad | \text{snd}(\langle \text{Expression} \rangle)
 \end{aligned}$$

Examples of programs in this language are

1. $\text{pair}\ (2,3)$
2. $\text{pair}\ (\text{pair}(7,9),3)$
3. $\text{pair}(\text{zero?}(4),11-x)$
4. $\text{snd}(\text{pair}\ (\text{pair}(7,9),3))$

The abstract syntax includes the following additional variants:

```

type expr =
...
| Pair of expr*expr
| Fst of expr
| Snd of expr

```

1. Specify the interpreter (i.e. its set of results and the new evaluation rules). You may assume that you have a product operation \times that computes the product of two sets.
2. Extend the implementation of eval_expr to handle the new language constructs. Are there any new errors?

Exercise 2.2.4. Consider another extension to LET with pairs. Its concrete syntax is:

$$\begin{aligned} \langle \text{Expression} \rangle &::= \text{pair}(\langle \text{Expression} \rangle, \langle \text{Expression} \rangle) \\ &\mid \text{unpair}(\langle \text{Identifier} \rangle, \langle \text{Identifier} \rangle) = \langle \text{Expression} \rangle \text{ in } \langle \text{Expression} \rangle \end{aligned}$$

Pairs are constructed in the same way as in Exercise 2.2.3. However, to eliminate pairs instead of *fst* and *snd* we now have *unpair*. The expression *unpair* (*x,y*)=*e1* *in* *e2* evaluates *e1*, makes sure it is a pair with components *v1* and *v2* and then evaluates *e2* in the extended environment where *x* is bound to *v1* and *y* to *v2*. Examples of programs in this extension are the first three examples in Exercise 2.2.3 and:

1. *unpair* (*x,y*) = *pair*(3, *pair*(5 , 12)) *in* *x* is a program that evaluates to *Ok* (*NumVal* 3).
2. The program *let* *x* = 34 *in* *unpair* (*y,z*)=*pair*(2,*x*) *in* *z* evaluates to *Ok* (*NumVal* 34).

The abstract syntax of this extension is:

```
type expr =
  ...
  | Pair of expr*expr
  | Unpair of string*string*expr*expr
```

[parser_plaf/lib/ast.ml](#)

Specify the interpreter (i.e. its evaluation rules) and then implement it.

Exercise 2.2.5. Consider the extension of LET with tuples. Its concrete syntax is that of LET together with the following new productions:

$$\begin{aligned} \langle \text{Expression} \rangle &::= \langle \langle \text{Expression} \rangle^{*(,)} \rangle \\ \langle \text{Expression} \rangle &::= \text{untuple} \langle \langle \text{Identifier} \rangle^{*(,)} \rangle = \langle \text{Expression} \rangle \text{ in } \langle \text{Expression} \rangle \end{aligned}$$

The $\langle \rangle^{*(,)}$ above the nonterminal indicates zero or more copies separated by commas. The angle brackets construct a tuple with the values of its arguments. An expression of the form *untuple* $\langle x_1, \dots, x_n \rangle = e_1$ *in* *e2* first evaluates *e1*, makes sure it is a tuple of *n* values, say *v1* to *vn*, and then evaluates *e2* in the extended environment where each identifier *x_i* is bound to *vi*. Examples of programs in this extension are:

1. $\langle 2, 3, 4 \rangle$
2. $\langle 2, 3, \text{zero?}(0) \rangle$
3. $\langle \langle 7, 9 \rangle, 3 \rangle$
4. $\langle \text{zero?}(4), 11-x \rangle$
5. *untuple* $\langle x, y, z \rangle = \langle 3, \langle 5, 12 \rangle, 4 \rangle$ *in* *x* evaluates to *Ok* (*NumVal* 3).
6. *let* *x* = 34 *in* *untuple* $\langle y, z \rangle = \langle 2, x \rangle$ *in* *z* evaluates to *Ok* (*NumVal* 34).

Specify the interpreter (i.e. its evaluation rules) and then implement it.

Exercise 2.2.6. Consider the following extension of LET with records. Its concrete syntax is given adding the following new productions to that of LET:

$$\begin{aligned} \langle \text{Expression} \rangle &::= \{ \{ \langle \text{Identifier} \rangle = \langle \text{Expression} \rangle \}^{+ (i)} \} \\ \langle \text{Expression} \rangle &::= \langle \text{Expression} \rangle . \langle \text{Identifier} \rangle \end{aligned}$$

Examples of programs in this extension are:

1. `{age=2; height=3}`
2. `let person = {age=2; height=3} in let student = {pers=person; cwid=10} in student`
3. `{age=2; height=3}.age`
4. `{age=2; height=3}.ages`
5. `{age=2; age=3}`

Assume that the expressed values of LET are extended so that now records of expressed values may be produced as a result of evaluating a program (see the examples above).

```
type exp_val =
  ...
  | RecordVal of (string*exp_val) list
```

The `expr` type encoding the AST is also extended:

```
type expr =
  ...
  | Record of (string*expr) list
  | Proj of expr*string
```

Specify the interpreter (i.e. its evaluation rules) and then implement it. Some examples of the result of evaluation of this extension are:

1. `{age=2; height=3}`
Should evaluate to `Ok (RecordVal [("age", NumVal 2); ("height", NumVal 3)])`.
2. `let person = {age=2; height=3} in let student = {pers=person; cwid=10} in student`
Should evaluate to `Ok (RecordVal [("pers", RecordVal [("age", NumVal 2); ("height", NumVal 3)]); ("cwid", NumVal 10)])`.
3. `{age=2; height=3}.age`
Should evaluate to `Ok (NumVal 2)`.
4. `{age=2; height=3}.ages`
Should evaluate to `Error "Field not found"`.
5. `{age=2; age=3}`
Should evaluate to `Error "Record has duplicate fields"`.

2.3 PROC

This section adds first-class functions to LET.

2.3.1 Concrete Syntax

Some examples of expressions in PROC are listed below:

```

1  let f = proc (x) { x-11 }
2  in (f (f 77))

4  (proc (f) { (f (f 77)) } proc (x) { x-11 })

6  let x = 2
   in let f = proc (z) { z-x }
   in (f 1)

10 let x = 2
   in let f = proc (z) { z-x }
12 in let x = 1
   in let g = proc (z) { z-x }
14 in (f 1) - (g 1)

```

The concrete syntax for PROC consists in adding two new productions to the grammar of the concrete syntax for LET:

```

<Expression> ::= <Number>
<Expression> ::= <Identifier>
<Expression> ::= <Expression> <BOP> <Expression>
<Expression> ::= zero?(<Expression>)
<Expression> ::= if <Expression> then <Expression> else <Expression>
<Expression> ::= let <Identifier> = <Expression> in <Expression>
<Expression> ::= (<Expression>)
<Expression> ::= proc(<Identifier>){<Expression>}
<Expression> ::= (<Expression>)<Expression>

<BOP>         ::= +|-|*|/

```

2.3.2 Abstract Syntax

```

1  type expr =
2  | Var of string
   | Int of int
4  | Add of expr*expr
   | Sub of expr*expr
6  | Mul of expr*expr
   | Div of expr*expr
8  | Let of string*expr*expr
   | IsZero of expr
10 | ITE of expr*expr*expr
   | Proc of string*texpr option*expr
12 | App of expr*expr

```

Note that the `Proc` constructor has three arguments. The first is the formal parameter and the last is the body of the function. The second parameter is an optional type annotation. It will play a role when we study type-checking. For now, values of type `expr` constructed using `Proc` will always have the form `Proc(id, None, e)`. For example, parsing the expression `let f=proc(x) x+1 in (f 3)` will produce the AST:

$$\begin{array}{c}
\frac{}{\text{Proc}(\text{id}, \text{e}), \rho \Downarrow (\text{id}, \text{e}, \rho)} \text{EProc} \\
\\
\frac{\text{e1}, \rho \Downarrow (\text{id}, \text{e}, \sigma) \quad \text{e2}, \rho \Downarrow w \quad \text{e}, \sigma \oplus \{\text{id} := w\} \Downarrow v}{\text{App}(\text{e1}, \text{e2}), \rho \Downarrow v} \text{EApp} \\
\\
\frac{\text{e1}, \rho \Downarrow v \quad v \notin \mathbb{CL}}{\text{App}(\text{e1}, \text{e2}), \rho \Downarrow \text{error}} \text{EAppErr}
\end{array}$$

Figure 2.7: Additional Evaluation rules for PROC (error propagation rules omitted)

```

AProg([],
  Let ("f", Proc ("x", None, Add (Var "x", Int 1)), App (Var "f", Int 3)))

```

2.3.3 Interpreter

2.3.3.1 Specification

Evaluation judgements for PROC are exactly the same as for LET except that now **the value resulting from evaluation of non-error computations, namely the expressed values, may either be an integer, a boolean or a closure**. A closure is a triple consisting of an identifier, an expression and an environment. All three sets, expressed values, closures and environments must be defined mutually recursively since they depend on each other:

$$\begin{aligned}
\text{EV} &:= \mathbb{Z} \cup \mathbb{B} \cup \mathbb{CL} \\
\mathbb{CL} &:= \{(\text{id}, \text{e}, \rho) \mid \text{e} \in \text{EXP}, \text{id} \in \mathbb{ID}, \rho \in \text{ENV}\} \\
\text{ENV} &:= \mathbb{ID} \rightarrow \text{EV}
\end{aligned}$$

The evaluation judgement for PROC reads:

$$\text{e}, \rho \Downarrow r$$

The evaluation rules include those of LET (see Figure 2.1) plus the additional rules given in Figure 2.7.

2.3.3.2 Implementation

To extend the interpreter for LET to PROC, we need to model closures and then extend `eval_expr`. Modeling closures as runtime values is easy since closures are simply triples consisting of an identifier, an expression and an environment:

```

type exp_val =
2 | NumVal of int
  | BoolVal of bool
4 | ProcVal of string*expr*env
and

```

```

6  env =
  | EmptyEnv
8  | ExtendEnv of string*exp_val*env

```

ds.ml

Now, for `eval_expr`, we add code for two new variants in the definition of `eval_expr`, namely `Proc(id,e)` and `App(e1,e2)`. Let us first analyze the former. Our first attempt might look something like this:

```

let rec eval_expr : expr -> exp_val ea_result =
2  fun e ->
  match e with
4  | Proc(id,e) ->
    return (ProcVal(id,e, en))

```

Evaluation of `Proc(id,e)` should produce a closure that includes both of `id` and `e`. It must also include the current environment, denoted `en` above. However, the identifier `en` is not in scope. Indeed, the current environment is passed around in the background by the helper functions for `ea_result`. We introduce a new helper function that reads the current environment and returns it as a result (*i.e.* `Ok env`, where `env` is the environment).

```

let lookup_env : env ea_result =
2  fun env -> Ok env

```

ds.ml

With this new function we can now implement the evaluator for `Proc(id,e)`:

```

let rec eval_expr : expr -> exp_val ea_result =
2  fun e ->
  match e with
4  | Proc(id,e) ->
    lookup_env >=> fun en ->
6    return (ProcVal(id,e,en))

```

interp.ml

Two alternative implementations for the `Proc(id,e)` case might be:

```

| Proc(id,e) ->
2  fun env -> return (ProcVal(id,e,env)) env

```

and

```

| Proc(id,e) ->
2  fun env -> Ok (ProcVal(id,e,env))

```

This last one is perhaps the least recommendable since the constructor `Ok` should best not be used outside `ds.ml`.

We next consider the case for `App(e1,e2)`. Evaluation of an application requires that we first evaluate `e1` and make sure it is a closure. If that is the case, then `clos` will be bound to a triple containing its three components (parameter, body and environment). We then evaluate `e2` and feed it to the helper function `apply_clos clos` (explained below), which does the rest of the job:

```

let rec eval_expr : expr -> exp_val ea_result =
2  fun e ->
  match e with
4  | App(e1,e2) ->

```

```

let rec apply_clos : string*expr*env -> exp_val ->
2   exp_val ea_result =
   fun (id,e,en) ev ->
4   return en >>+
   extend_env id ev >>+
6   eval_expr e
and
8   eval_expr : expr -> exp_val ea_result =
   fun e ->
10  match e with
   | Proc(id,e) ->
12    lookup_env >>= fun en ->
    return (ProcVal(id,e,en))
14  | App(e1,e2) ->
    eval_expr e1 >>=
16    clos_of_procVal >>= fun clos ->
    eval_expr e2 >>=
18    apply_clos clos

```

interp.ml

Figure 2.8: Interpreter for PROC

```

6   eval_expr e1 >>=
   clos_of_procVal >>= fun clos ->
   eval_expr e2 >>=
8   apply_clos clos

```

The function `clos_of_procVal` is similar to `int_of_numVal` from Listing 2.2. Its code is:

```

let clos_of_procVal : exp_val -> (string*expr*env) ea_result =
2   fun ev ->
   match ev with
4   | ProcVal(id,body,en) -> return (id,body,en)
   | _ -> error "Expected a closure!"

```

ds.ml

The function `apply_clos` sets `en` to be the new current environment and then extends it with the assignment of `id` to `ev`. Under this extended environment, it proceeds with the evaluation of the body of the closure, namely `e`.

```

let rec apply_clos : string*expr*env -> exp_val -> exp_val ea_result =
2   fun (id,e,en) ev ->
   return en >>+
4   extend_env id ev >>+
   eval_expr e

```

In passing we mention that the tuple type constructor has higher precedence than the function type constructor. Consequently, there is no need to place the type expression `string*expr*env` between parenthesis.

Listing 2.8 summarizes the code described above for procedures and applications.

2.3.4 Dynamic Scoping

If we remove the line below, then we implement dynamic scoping:

```

let rec apply_clos : string*expr*env -> exp_val -> exp_val ea_result =
2   fun (id,e,en) ev ->
    return en ->>+
4   (extend_env id ev >>+
    eval_expr e)

```

Indeed, in this case the environment that is extended by `extend_env id a` is the current environment and not the one that was saved in the closure. Here are some examples of executing programs in this variant of PROC:

```

# interp "
2 let f = proc (x) { if zero?(x) then 1 else x*(f (x-1)) }
  in (f 6) ";;
4 - : Ds.exp_val Ds.result = Ds.Ok (Ds.NumVal 720)

6 # interp "
  let f = proc (x) { x+a }
8  in let a=2
    in (f 2) ";;
10 - : Ds.exp_val Ds.result = Ds.Ok (Ds.NumVal 4)

12 # interp "
    let f= let a=2 in proc(x) { x+a}
14 in (f 2) ";;
    - : Ds.exp_val Ds.result = Ds.Error "a not found!"

```

utop

2.4 Exercises

Exercise 2.4.1. Write a grammar derivation to show that `let f = proc (x) { x-11 } in (f 77)` is a valid program in PROC.

Exercise 2.4.2. Write down the parse tree for the expression `let pred = proc(x) { x-1 } in (pred 5)`.

Exercise 2.4.3. Write down the result of evaluating the expressions below. Depict the full details of the closure, including the environment. Use the tabular notation seen in class to depict the environment.

- `proc (x) { x-11 }`
- `proc (x) { let y=2 in x }`
- `let a=1 in proc (x) { x }`
- `let a=1 in let b=2 in proc (x) { x }`
- `let f=(let b=2 in proc (x) { x }) in f`
- `proc (x) { proc (y) { x-y } }`

Exercise 2.4.4. Depict the environment extant at the breakpoint (signalled with the *debug* expression):

```

let a=1
in let b=2
in let c=proc (x) { x }
in debug((c b))

```

Exercise 2.4.5. *Depict the environment extant at the breakpoint:*

```

let a=1
in let b=2
in let c = proc (x) { debug(proc (y) { x-y }) }
in (c b)

```

Exercise 2.4.6. *Depict the environment extant at the breakpoint:*

```

let x=2
in let y=proc (d) { x }
in let z=proc(d) { x }
in debug(3)

```

Exercise 2.4.7. *The result of evaluating the following expression is $\text{Ok } (\text{NumVal } 4)$. Verify this.*

```

let a = 3
in let p = proc (z) { z+a }
in let f = proc (x) { (p 1) }
in let a = 6
in (f 2)

```

Modify the interpreter so that it uses dynamic scoping rather than static scoping. Then evaluate the above expression again. What value does it return?

Exercise 2.4.8 (\diamond). Use the “higher-order” trick of self-application to implement the mutually recursive definitions of *even* and *odd* in PROC:

$$\begin{aligned} \text{even}(0) &= \text{true} \\ \text{even}(n) &= \text{odd}(n-1) \\ \\ \text{odd}(0) &= \text{false} \\ \text{odd}(n) &= \text{even}(n-1) \end{aligned}$$

Exercise 2.4.9 (\diamond). Use the “higher-order” trick of self-application to implement a function *pbt* that given a value v and a height h builds a perfect binary tree constructed out of pairs and that has v in the leaves and has height v . For example $((\text{pbt } 2) 3)$ should produce

```

PairVal
(PairVal
  (PairVal (PairVal (NumVal 2, NumVal 2),
    PairVal (NumVal 2, NumVal 2)),
    PairVal (PairVal (NumVal 2, NumVal 2),
      PairVal (NumVal 2, NumVal 2))),
  PairVal
    (PairVal (PairVal (NumVal 2, NumVal 2),
      PairVal (NumVal 2, NumVal 2)),
      PairVal (PairVal (NumVal 2, NumVal 2),
        PairVal (NumVal 2, NumVal 2))))

```


2.5 REC

Our language unfortunately does not support recursion⁴. The following attempt at defining factorial and then applying it to compute factorial of 5 fails. The reason is that x is not visible in the body of the `proc`.

```
1 let f =  
2   proc (x) {  
3     if zero?(x)  
4       then 1  
5       else x*(f (x-1)) }  
6 in (f 5)
```

In order to verify this, evaluate the following expression:

```
1 let f =  
2   proc (x) {  
3     debug(if zero?(x)  
4       then 1  
5       else x*(f (x-1))) }  
6 in (f 5)
```

Note that the environment in the closure for x does not include a reference to x itself. The next language we shall look at, namely REC, includes a new programming abstraction that allows us to define recursive functions. In REC we will write:

```
1 letrec fact(x) =  
2   if zero?(x)  
3   then 1  
4   else x * (fact (x-1))  
in (fact 5)
```

REC also supports mutually recursive function declarations such as:

```
1 let true = zero?(0)  
in let false = zero?(1)  
3 in letrec  
4   even(x) = if zero?(x) then true else (odd (x-1))  
5   odd(x) = if zero?(x) then false else (even (x-1))  
in (odd 99)
```

⁴See exercises 2.4.8 and 2.4.9 on the “higher-order” trick though.

2.5.1 Concrete Syntax

```

<Expression> ::= <Number>
<Expression> ::= <Identifier>
<Expression> ::= <Expression> <BOp> <Expression>
<Expression> ::= zero?(<Expression>)
<Expression> ::= if <Expression> then <Expression> else <Expression>
<Expression> ::= let <Identifier> = <Expression> in <Expression>
<Expression> ::= (<Expression>)
<Expression> ::= proc(<Identifier>){<Expression>}
<Expression> ::= (<Expression> <Expression>)
<Expression> ::= letrec {<Identifier>(<Identifier>) = <Expression>}+ in <Expression>

<BOp> ::= + | - | * | /

```

Note that the curly braces in the last grammar production for `<Expression>` are not terminals, they simply indicate that the sequence of terminals and non-terminals '`<Identifier>(<Identifier>) = <Expression>`' may occur once or more.

2.5.2 Abstract Syntax

```

type expr =
  | Var of string
  | Int of int
  | Add of expr*expr
  | Sub of expr*expr
  | Mul of expr*expr
  | Div of expr*expr
  | Let of string*expr*expr
  | IsZero of expr
  | ITE of expr*expr*expr
  | Proc of string*texpr option*expr
  | App of expr*expr
  | Letrec of rdec*expr
and
  rdec = (string*string*texpr option*texpr option*expr) list

```

[parser_plaf/lib/ast.ml](#)

For example, the result of parsing the expression:

```

1 letrec fact(x) =
    if zero?(x)
3   then 1
    else x * (fact (x-1))
5 in (fact 5)

```

is the AST:

```

1 AProg ([,
  Letrec
3   ([("fact", "x", None, None,
      ITE (IsZero (Var "x"), Int 1,
5      Mul (Var "x", App (Var "fact", Sub (Var "x", Int 1)))))],
  App (Var "fact", Int 5))

```

$$\begin{array}{c}
\frac{\mathbf{e2}, \rho \oplus \{\mathbf{id} := (\mathbf{par}, \mathbf{e1}, \rho)^r\} \Downarrow v}{\mathbf{Letrec}([\mathbf{id}, \mathbf{par}, \mathbf{e1}], \mathbf{e2}), \rho \Downarrow v} \text{ELetRec} \\
\\
\frac{\rho(\mathbf{id}) = (\mathbf{par}, \mathbf{e}, \sigma)^r}{\mathbf{Var}(\mathbf{id}), \rho \Downarrow (\mathbf{par}, \mathbf{e}, \sigma \oplus \{\mathbf{id} := (\mathbf{par}, \mathbf{e}, \sigma)^r\})} \text{EVarLetRec}
\end{array}$$

Figure 2.9: Additional evaluation rules for REC

The arguments `None` may be ignored for now. They indicate there are no typing annotations; we shall consider typing annotations in Chapter 4. Also, the syntax for `letrec` supports the definition of mutually recursive functions (hence the reason for `list` in the type definition `rdecs`), however for now we will only be using examples where a single recursive function is declared.

2.5.3 Interpreter

Recursive functions will be represented as special closures called **recursion closures**. Later we will look at another implementation involving circular environments. A **recursion closure** is a closure with a tag “*r*” to distinguish it from a standard closure, written $(\mathbf{id}, \mathbf{e}, \rho)^r$, where $e \in \text{EXP}$, $id \in \text{ID}$ and $\rho \in \text{ENV}$. The set of all recursion closures is denoted RCL :

$$\begin{aligned}
\text{ENV} &:= \text{ID} \rightarrow (\text{EV} \cup \text{RCL}) \\
\text{EV} &:= \text{Z} \cup \text{B} \cup \text{CL} \\
\text{CL} &:= \{(\mathbf{id}, \mathbf{e}, \rho) \mid e \in \text{EXP}, id \in \text{ID}, \rho \in \text{ENV}\} \\
\text{RCL} &:= \{(\mathbf{id}, \mathbf{e}, \rho)^r \mid e \in \text{EXP}, id \in \text{ID}, \rho \in \text{ENV}\}
\end{aligned}$$

Note that recursion closures are not expressed values. We cannot write a program that, when evaluated, returns a recursion closure. They are an auxiliary device for defining evaluation of recursive programs. More precisely, recursive function definitions will be stored as recursion closures. However, lookup of recursive functions will produce standard closures, the latter being computed on the fly.

2.5.3.1 Specification

The set of results is the same as in PROC:

$$\mathbb{R} := \text{EV} \cup \{\text{error}\}$$

Evaluation judgements for REC are the same as for PROC:

$$\mathbf{e}, \rho \Downarrow r$$

where $r \in \mathbb{R}$. Evaluation rules for REC are those of PROC together with the ones in Figure 2.9. Two new evaluation rules are added to those of PROC to obtain REC. The rule `ELetRec` creates a recursion closure and adds it to the current environment ρ and then continues with the evaluation of `e2`. The rule `EVarLetRec` does lookup of identifiers that refer to previously declared recursive functions. Upon finding the corresponding recursion closure in the current environment, it creates a new closure and returns it. Note that the newly created closure includes an environment that has a reference to \mathfrak{x} itself.

2.5.3.2 Implementation

Recursion closures are implemented by adding a new constructor to the type `expr`, namely `ExtendEnvRec` below:

```

type exp_val =
2   | NumVal of int
   | BoolVal of bool
4   | ProcVal of string*expr*env
and
6   env =
   | EmptyEnv
8   | ExtendEnv of string*exp_val*env
   | ExtendEnvRec of string*string*expr*env

```

ds.ml

Note that the arguments of `ExtendEnvRec(id,par,body,env)` are four: the name of the recursive function being defined `id`, the name of the formal parameter `par`, the body of the recursive function `body`, and the rest of the environment `env`. If we consider the environment $\rho \oplus \{id := (par, e1, \rho)\}$ in the rule `ELetRec` of Figure 2.9, it would seem we are missing an argument. Indeed, the operator “ \oplus ” in the evaluation rule is modeled by the `ExtendEnvRec` constructor in our implementation. However, there is no need to store ρ in our implementation since it is just the tail of the environment.

Next we need an operation similar to `extend_env` but that adds a new recursion closure to the environment:

```

let extend_env_rec : string -> string -> expr -> env ea_result =
2   fun id par body ->
     fun env -> Ok (ExtendEnvRec(id,par,body,env))

```

ds.ml

In addition, we need to update the implementation of `apply_env` so that it deals with lookup of recursive functions, thus correctly implementing `EVarLetRec`. This involves adding a new clause (see code highlighted below):

```

let rec apply_env : string -> exp_val ea_result =
2   fun id ->
     fun env ->
4     match env with
     | EmptyEnv -> Error (id^" not found!")
6     | ExtendEnv(v,ev,tail) ->
         if id=v
8         then Ok ev
         else apply_env id tail
10    | ExtendEnvRec(v,par,body,tail) ->
         if id=v
12    then Ok (ProcVal (par,body,env))
         else apply_env id tail

```

ds.ml

Regarding the code for the interpreter itself, we need only add a new clause, namely the one for `Letrec(id,par,e1,e2)`:

```

2   | Letrec([(id,par,_,_,e1)],e2) ->
     extend_env_rec id par e1 >>+
     eval_expr e2

```

interp.ml

Exercise 2.5.1. Evaluate the following expressions in utop:

```
1.
1 utop # interp "
  let one=1
3 in letrec fact(x) =
      if zero?(x)
5       then one
      else x * (fact (x-1))
7 in debug((fact 6))" ;;
```

```
2.
1 utop # interp "
  let one=1
3 in letrec fact(x) =
      debug(if zero?(x)
5         then one
         else x * (fact (x-1)))
7 in (fact 6)" ;;
```

```
3.
1 utop # interp "
  let one=1
3 in letrec fact(x) =
      if zero?(x)
5       then one
      else x * (fact (x-1))
7 in fact" ;;
```

Exercise 2.5.2 (\diamond). Consider the following functions in OCaml:

```
1 let rec add n m =
  match n with
3   | 0 -> m
   | n' -> 1 + add (n'-1) m
5
11 let rec append l1 l2 =
  match l1 with
7   | [] -> l2
   | h::t -> h :: append t l2
9
11 let rec map l f =
  match l with
13  | [] -> []
   | h::t -> (f h) :: map t f
15
17 let rec filter l p =
  match l with
19  | [] -> []
   | h::t ->
21     if p h
     then h :: filter t p
     else filter t p
23
```

```

25 let rec foldr l f a =
    match l with
27 | [] -> a
    | h::t -> f h (foldr t f a)

```

Code them in the extension of REC of Exercise 2.4.10. For example, here is the code for `add`:

```

# interp "
2 letrec add(n) = proc (m) {
    if zero?(n)
4         then m
    else 1 + ((add (n-1)) m) }
6 in ((add 2) 3)";;
- : exp_val Rec.Ds.result = Ok (NumVal 5)

```

utop

Exercise 2.5.3 (\diamond). Consider the following expression in REC

```

let z = 0
in let prod = proc (x) { proc (y) { x*y }}
in letrec f(n) = if zero?(n) then 1 else ((prod n) (f (n-1)))
in (f 10)

```

A *debug* instruction was placed somewhere in the code and it produced the environments below. Where was it placed? Identify and signal (see instructions below) the location for each of the three items below. Note that there may be more than one solution for each item, it suffices to supply just one.

1.

```

>>Environment:
z:=NumVal 0,
prod:=ProcVal (x,Proc(y,Mul(Var x,Var y))),z:=NumVal 0),
f:=Rec(n,IfThenElse(Zero?(Var n),Int 1,
    App(App(Var prod,Var n),App(Var f,Sub(Var n,Int 1))))),
n:=NumVal 0

```

Draw a box around the argument of *debug*:

```

let z = 0
in let prod = proc (x) { proc (y) { x*y }}
in letrec f(n) = if zero?(n) then 1 else ((prod n) (f (n-1)))
in (f 10)

```

2.

```

>>Environment:
z:=NumVal 0

```

Draw a box around the argument of *debug*:

```

let z = 0
in let prod = proc (x) { proc (y) { x*y }}
in letrec f(n) = if zero?(n) then 1 else ((prod n) (f (n-1)))
in (f 10)

```

3.

```
>>Environment:
z:=NumVal 0,
x:=NumVal 10
```

Draw a box around the argument of debug:

```
let z = 0
in let prod = proc (x) { proc (y) { x*y }}
in letrec f(n) = if zero?(n) then 1 else ((prod n) (f (n-1)))
in (f 10)
```


Chapter 3

Imperative Programming

3.1 Mutable Data Structures in OCaml

This section discusses some OCaml language features that all data to be updated in-place. We will introduce references, arrays and mutable record fields.

3.1.1 A counter object

```
2 type counter =  
4   { inc: unit -> unit;  
    dec: unit -> unit;  
    read : unit -> int }
```

```
2 let c =  
4   let state = ref 0 in  
   { inc = (fun () -> state := !state+1);  
     dec = (fun () -> state := !state-1);  
     read = (fun () -> !state) }
```

```
1 # c.read ();;  
- : int = 0  
3 # c.inc ();;  
- : unit = ()  
5 # c.read ();;  
- : int = 1  
7 # c.dec ();;  
- : unit = ()  
9 # c.read ();;  
- : int = 0
```

utop

3.1.2 A stack object

```

type stack =
2   { push : int -> unit;
    pop  : unit -> int;
4    top  : unit -> int};;

let s = let state = ref []
2       in { push = (fun i -> state := i :: !state);
            pop  = (fun () -> let temp = List.hd !state
4                          in state := List.tl !state; temp);
            top  = (fun () -> List.hd !state)}

1  # s.push 1;;
   - : unit = ()
3  # s.push 2;;
   - : unit = ()
5  # s.pop ();;
   - : int = 2
7  # s.top ();;
   - : int = 1

```

utop

3.2 EXPLICIT-REFS

The following is an extension of REC.

3.2.1 Concrete Syntax

Examples of expressions in EXPLICIT-REFS:

```

newref(2)
2
let a=newref(2)
4 in a

6 let a=newref(2)
  in deref(a)

8
10 let a=newref(2)
   in setref(a,deref(a)+1)

12 let a=newref(2)
   in begin
14       setref(a,deref(a)+1);
       deref(a)
16   end

18 let g =
   let counter = newref(0)
20   in proc (d) {
       begin
22       setref(counter, deref(counter)+1);
       deref(counter)
24   end

```

```

    }
in (g 11) - (g 22)

```

```

⟨Expression⟩ ::= ⟨Number⟩
⟨Expression⟩ ::= ⟨Identifier⟩
⟨Expression⟩ ::= ⟨Expression⟩⟨BOp⟩⟨Expression⟩
⟨Expression⟩ ::= zero?(⟨Expression⟩)
⟨Expression⟩ ::= if ⟨Expression⟩ then ⟨Expression⟩ else ⟨Expression⟩
⟨Expression⟩ ::= let ⟨Identifier⟩ = ⟨Expression⟩ in ⟨Expression⟩
⟨Expression⟩ ::= (⟨Expression⟩)
⟨Expression⟩ ::= proc(⟨Identifier⟩){⟨Expression⟩}
⟨Expression⟩ ::= (⟨Expression⟩⟨Expression⟩)
⟨Expression⟩ ::= letrec⟨Identifier⟩(⟨Identifier⟩)=⟨Expression⟩in⟨Expression⟩
⟨Expression⟩ ::= newref(⟨Expression⟩)
⟨Expression⟩ ::= deref(⟨Expression⟩)
⟨Expression⟩ ::= setref(⟨Expression⟩,⟨Expression⟩)
⟨Expression⟩ ::= begin ⟨Expression⟩*(;) end

⟨BOp⟩ ::= + | - | * | /

```

The notation $*(;)$ above the nonterminal $\langle \text{Expression} \rangle$ in the production for `begin/end` indicates zero or more expressions separated by semi-colons.

3.2.2 Abstract Syntax

```

type expr =
2   | Var of string
   | Int of int
4   | Add of expr*expr
   | Sub of expr*expr
6   | Mul of expr*expr
   | Div of expr*expr
8   | Let of string*expr*expr
   | IsZero of expr
10  | ITE of expr*expr*expr
   | Proc of string*expr
12  | App of expr*expr
   | Letrec of string*string*expr*expr
14  | NewRef of expr
   | DeRef of expr
16  | SetRef of expr*expr
   | BeginEnd of expr list
18  | Debug of expr

```

3.2.3 Interpreter

3.2.3.1 Specification

We assume given a set of (symbolic) memory locations \mathbb{L} . We write ℓ, ℓ_i for memory locations. A heap or **store** is a partial function from memory locations to expressed values. The set of

$$\begin{array}{c}
\frac{\mathbf{e}, \rho, \sigma \Downarrow v, \sigma' \quad \ell \notin \text{dom}(\sigma')}{\text{NewRef}(\mathbf{e}), \rho, \sigma \Downarrow \ell, \sigma' \oplus \{\ell := v\}} \text{ENewRef} \\
\\
\frac{\mathbf{e}, \rho, \sigma \Downarrow v, \sigma' \quad v \in \mathbb{L} \quad v \in \text{dom}(\sigma')}{\text{DeRef}(\mathbf{e}), \rho, \sigma \Downarrow \sigma'(v), \sigma'} \text{EDeRef} \\
\\
\frac{\mathbf{e}, \rho, \sigma \Downarrow v, \sigma' \quad v \notin \mathbb{L}}{\text{DeRef}(\mathbf{e}), \rho, \sigma \Downarrow \text{error}, \sigma'} \text{EDeRefErr1} \quad \frac{\mathbf{e}, \rho, \sigma \Downarrow v, \sigma' \quad v \in \mathbb{L} \quad v \notin \text{dom}(\sigma')}{\text{DeRef}(\mathbf{e}), \rho, \sigma \Downarrow \text{error}, \sigma'} \text{EDeRefErr2} \\
\\
\frac{\mathbf{e1}, \rho, \sigma \Downarrow v, \sigma' \quad v \in \mathbb{L} \quad \mathbf{e2}, \rho, \sigma' \Downarrow w, \sigma''}{\text{SetRef}(\mathbf{e1}, \mathbf{e2}), \rho, \sigma \Downarrow \text{unit}, \sigma'' \oplus \{v := w\}} \text{ESetRef} \quad \frac{\mathbf{e1}, \rho, \sigma \Downarrow v, \sigma' \quad v \notin \mathbb{L}}{\text{SetRef}(\mathbf{e1}, \mathbf{e2}), \rho, \sigma \Downarrow \text{error}, \sigma'} \text{ESetRefErr} \\
\\
\frac{n > 0 \quad (\mathbf{ei}, \rho, \sigma_i \Downarrow v_i, \sigma_{i+1})_{i \in 1..n}}{\text{BeginEnd}([\mathbf{e1}; \dots; \mathbf{en}]), \rho, \sigma_1 \Downarrow v_n, \sigma_{n+1}} \text{EBeginEndNE} \\
\\
\frac{}{\text{BeginEnd}([]), \rho, \sigma \Downarrow \text{unit}, \sigma} \text{EBeginEndE}
\end{array}$$

Figure 3.1: Evaluation rules for EXPLICIT-REFS (error propagation rules omitted)

stores is denoted \mathbb{S} :

$$\mathbb{S} := \mathbb{L} \rightarrow \mathbb{EV}$$

where the set of expressed values includes locations:

$$\mathbb{EV} := \mathbb{Z} \cup \mathbb{B} \cup \mathbb{U} \cup \mathbb{C} \cup \mathbb{L}$$

Also among expressed values we find $\mathbb{U} := \{\text{unit}\}$. This new value will be explained below, when we describe the evaluation rules for EXPLICIT-REFS.

Evaluation judgements in EXPLICIT-REFS take the following form, where \mathbf{e} is an expression, ρ and environment, σ the initial store, r the result and σ' the final store

$$\mathbf{e}, \rho, \sigma \Downarrow r, \sigma'$$

Note that the result of evaluating an expression now returns both a result and an updated store. The evaluation rules for EXPLICIT-REFS are given in Figure 3.1. The rule ESetRef and ESetRefErr describe the behavior of assignment. Notice that an assignment such as SetRef($\mathbf{e1}, \mathbf{e2}$) is evaluated to cause an effect, namely update the contents of the location obtained from evaluating $\mathbf{e1}$ with the value obtained from evaluating $\mathbf{e2}$. We do not expect to get any meaningful value back. However, all expressions have to denote a value. As a consequence, we use a new expressed value *unit*, as the expressed value returned by an assignment.

3.2.3.2 Implementing Stores

The implementation of the evaluator for EXPLICIT-REFS requires that we first implement stores. Since a store is a mutable data structure we will use OCaml arrays. The following interface file declares the types of the values in the public interface of the store. These values include a parametric type constructor `Store.t`, the type of the store itself and multiple functions.

```

open Ds
2 type 'a t

4 val empty_store : int -> 'a -> 'a t
  val get_size : 'a t -> int
6 val new_ref : 'a t -> 'a -> int
  val deref : 'a t -> int -> 'a ea_result
8 val set_ref : 'a t -> int -> 'a -> unit ea_result
  val string_of_store : ('a -> string) -> 'a t -> string

```

store.mli

These operations are:

- `empty_store n v` returns a store of size `n` where each element is initialized to `v`
- `get_size s` returns the number of elements in the store.
- `new_ref s v` stores `v` in a fresh location and returns the location.
- `deref s l` returns the contents of location `l`, prefixed by `Some`, in the store `s`. This operation fails, returning `None`, if the location is out of bounds.
- `set_ref s l v` updates the contents of `l` in `s` with `v`. It fails, returning `None`, if the index is out of bounds.
- `string_of_store to_str s` returns a string representation of `s` resulting from applying `to_str` to each element.

Each of the above operations implemented in `store.ml`.

```

open Ds
2
3 type 'a t = { mutable data: 'a array; mutable size: int }
4 (* data is declared mutable so the store may be resized *)
5
6 let empty_store : int -> 'a -> 'a t =
7   fun i v -> { data=Array.make i v; size=0 }
8
9 let get_size : 'a t -> int =
10   fun st -> st.size
11
12 let enlarge_store : 'a t -> 'a -> unit =
13   fun st v ->
14     let new_array = Array.make (st.size*2) v
15     in Array.blit st.data 0 new_array 0 st.size;
16     st.data<-new_array
17
18 let new_ref : 'a t -> 'a -> int =
19   fun st v ->

```

```

20   if Array.length (st.data)=st.size
21   then enlarge_store st v
22   else ();
23   begin
24       st.data.(st.size)<-v;
25       st.size<-st.size+1;
26       st.size-1
27   end
28
29
30   let deref : 'a t -> int -> 'a ea_result =
31       fun st l ->
32           if l>=st.size
33           then error "Index out of bounds"
34           else return (st.data.(l))
35
36   let set_ref : 'a t -> int -> 'a -> unit ea_result =
37       fun st l v ->
38           if l>=st.size
39           then error "Index out of bounds"
40           else return (st.data.(l)<-v)
41
42   let rec take n = function
43       | [] -> []
44       | x::xs when n>0 -> x::take (n-1) xs
45       | _ -> []
46
47   let string_of_store' f st =
48       let ss = List.mapi (fun i x -> string_of_int i^"->"^f x) @@ take st.size @@ Array.to_list st.
49       in
50       String.concat ",\n" ss
51
52   let string_of_store f st =
53       match st.size with
54       | 0 -> ">>Store:\nEmpty"
55       | _ -> ">>Store:\n" ^ string_of_store' f st

```

store.ml



In OCaml, every .ml file is wrapped into a module. Modules package together related definitions and help provide consistent namespaces. For example, `store.ml` will be wrapped in a module called `Store`. Modules can provide not just functions but also types and submodules, among others. By default, all definitions provided in a module are accessible. Through interface files one may restrict the definitions that are accessible. For example, the `store.mli` file above, lists the definitions that are to be made accessible within the module `Store`.

3.2.3.3 Implementation

We could now follow the ideas we developed for environments and have stores threaded for us behind the scenes. This would lead to a similar extension of our current result type `ea_result` so that it also abstracts over the store. Also, the updated store would have to be returned. Thus the return type result would have to be updated to return a pair consisting of the updated store

and the result itself¹. However, in order to keep things simple and since the concept of threading behind the scenes has already been introduced via environments, we choose to hold the store in a top-level or global variable `g_store`.

```
let g_store = Store.empty_store 20 (NumVal 0)
```

interp.ml

`g_store` denotes a store of size 20, whose values have arbitrarily been initialized to `NumVal 0`.

Next we consider the new expressed values, namely symbolic locations and unit. Locations will be denoted by an integer wrapped inside a `RefVal` constructor. For example, `RefVal 7` is a pointer to memory location 7.

```
type exp_val =
2 | NumVal of int
  | BoolVal of bool
4 | ProcVal of string*expr*env
  | UnitVal
6 | RefVal of int
```

ds.ml

Next we move on to the interpreter, only addressing the new variants.

```
let rec eval_expr : expr -> exp_val ea_result =
2 fun e ->
  match e with
4 | NewRef(e) ->
    eval_expr e >>= fun ev ->
    return (RefVal (Store.new_ref g_store ev))
6 | DeRef(e) ->
    eval_expr e >>=
    int_of_refVal >>= fun l ->
    Store.deref g_store l
8 | SetRef(e1,e2) ->
    eval_expr e1 >>=
    int_of_refVal >>= fun l ->
    eval_expr e2 >>= fun ev ->
    Store.set_ref g_store l ev >>= fun _ ->
    return UnitVal
16 | BeginEnd([]) ->
    return UnitVal
18 | BeginEnd(es) ->
    sequence (List.map eval_expr es) >>= fun l ->
    return (List.hd (List.rev l))
20 | Debug(_e) ->
    string_of_env >>= fun str_env ->
    let str_store = Store.string_of_store string_of_expval g_store
    in (print_endline (str_env^"\n"^str_store);
    error "Debug called")
26 | _ -> failwith ("Not implemented: "^string_of_expr e)
```

interp.ml

¹This handling of the store in the background, including its auxiliary data types, is known as a state monad. Thus we would end up with a combination of error, reader, and state monads. Combining monads can be done through monad transformers.

3.2.4 Extended Example: Encoding Objects

EXPLICIT-REFS with records

```

1 let c = let s = newref(0)
2   in
3     {
4       inc = proc (d) { setref(s,deref(s)+d) };
5       read = proc (x) { deref(s) };
6       reset = proc (d) { setref(s,0) }
7     }
8 in begin
9   (c.inc 1);
10  (c.inc 2);
11  (c.read 0)
12 end

letrec self(s) =
2   { inc = proc (d) { setref(s,deref(s)+d) };
3     read = proc (x) { deref(s) };
4     reset = proc (d) {
5       let current = ((self s).read 0)
6       in ((self (s)).inc (-current))}
7   }
8 in let new_counter = proc(init) {
9   let s = newref(init)
10  in (self s)
11 }
12 in let c= (new_counter 0)
13 in begin
14   (c.inc 1);
15   (c.inc 2);
16   (c.reset 0);
17   (c.read 0)
18 end

```

3.3 IMPLICIT-REFS

The following is an extension of REC.

3.3.1 Concrete Syntax

Examples of expressions in IMPLICIT-REFS

```

1 let x=2
2 in begin
3   set x=3;
4   x
5 end

6 let x=2
7 in let y=x+1
8 in begin
9   set y=y+1;
10

```



```

      y
12 end
14 let x=2
in let f = proc (n) { begin set x=x+1; 1 end }
16 in let g = proc (n) { begin set x=x+1; 2 end }
in begin
18   (f 0)+(g 0);
   x
20 end

```

```

⟨Expression⟩ ::= ⟨Number⟩
⟨Expression⟩ ::= ⟨Identifier⟩
⟨Expression⟩ ::= ⟨Expression⟩⟨BOp⟩⟨Expression⟩
⟨Expression⟩ ::= zero?(⟨Expression⟩)
⟨Expression⟩ ::= if ⟨Expression⟩ then ⟨Expression⟩ else ⟨Expression⟩
⟨Expression⟩ ::= let ⟨Identifier⟩ = ⟨Expression⟩ in ⟨Expression⟩
⟨Expression⟩ ::= (⟨Expression⟩)
⟨Expression⟩ ::= proc(⟨Identifier⟩){⟨Expression⟩}
⟨Expression⟩ ::= (⟨Expression⟩⟨Expression⟩)
⟨Expression⟩ ::= letrec {⟨Identifier⟩(⟨Identifier⟩) = ⟨Expression⟩}+ in ⟨Expression⟩
⟨Expression⟩ ::= set ⟨Identifier⟩ = ⟨Expression⟩
⟨Expression⟩ ::= begin ⟨Expression⟩+(;) end

⟨BOp⟩ ::= +|-|*|/

```

3.3.2 Abstract Syntax

```

type expr =
2  | Var of string
  | Int of int
4  | Add of expr*expr
  | Sub of expr*expr
6  | Mul of expr*expr
  | Div of expr*expr
8  | Let of string*expr*expr
  | IsZero of expr
10 | ITE of expr*expr*expr
  | Proc of string*expr
12 | App of expr*expr
  | Letrec of rdec*expr
14 | Set of string*expr
  | BeginEnd of expr list
16 | Debug of expr
and
18 rdec = (string*string*texpr option*texpr option*expr) list

```

3.3.3 Interpreter

3.3.3.1 Specification

Since in IMPLICIT-REFS all identifiers are mutable, the environment will map all identifiers to locations in the store. Evaluation judgements in IMPLICIT-REFS take the following form, where e is an expression, ρ and environment, σ the initial store, r the result and σ' the final store

$$e, \rho, \sigma \Downarrow r, \sigma'$$

As mentioned, ρ maps identifiers to locations, it **no longer** maps them to expressed values. Hence identifier lookup now has to lookup the location first in the environment and then access the contents in the store. This is exactly what the rule EVar states:

$$\frac{\sigma(\rho(\text{id})) = v}{\text{Var}(\text{id}), \rho, \sigma \Downarrow v, \sigma} \text{EVar}$$

Indeed, $\rho(\text{id})$ denotes a location whose contents is looked up in the store σ . If $\rho(\text{id})$ is not a valid location, then an error is returned, as described by rule EVarErr. The full set of evaluation rules are given in Figure 3.2.

3.3.3.2 Implementation

We address the implementation of the evaluator. For now we ignore `Letrec` and then take it up later. Instead we focus on the `App(e1,e2)` case, which needs some minor updating, and also on the new variants.

Regarding the `App(e1,e2)` case, we need to slightly modify the `apply_clos` function. We briefly recall the code for `apply_clos` as implemented in the PROC (Figure 2.8):

```

let rec apply_clos : string*expr*env -> exp_val -> exp_val ea_result =
2   fun (id,e,en) ev ->
    return en >>+
4   (extend_env id ev >>+
    eval_expr e)

```

Note that evaluation of the body requires extending the environment with a new key-value pair, namely (id, ev) , where ev is the expressed value supplied as argument. Environments no longer map identifiers to expressed values, but to locations. So we first need to allocate ev in the store in a fresh location l and then extend the environment with the key-value pair (id, l) . The updated code for `apply_clos` is given below.

```

let rec apply_clos : string*expr*env -> exp_val -> exp_val ea_result =
2   fun (id,e,en) ev ->
    return en >>+
4   (extend_env id (RefVal (Store.new_ref g_store ev)) >>+
    eval_expr e)

```

We now address the new cases (and the ones we need to modify) for the interpreter:

```

1   | Var(id) ->
    apply_env id >>=
3   int_of_refVal >>= (* make sure id is mapped to a location *)
    Store.deref g_store (* if so, dereference it *)

```

$$\begin{array}{c}
\frac{\sigma(\rho(\text{id})) = v}{\text{Var}(\text{id}), \rho, \sigma \Downarrow v, \sigma} \text{EVar} \quad \frac{\rho(\text{id}) \notin \mathbb{L} \text{ or } \rho(\text{id}) \notin \text{dom}(\sigma)}{\text{Var}(\text{id}), \rho, \sigma \Downarrow \text{error}, \sigma} \text{EVarErr} \\
\\
\frac{\text{e1}, \rho, \sigma_1 \Downarrow w, \sigma_2 \quad \ell \notin \text{dom}(\sigma_2) \quad \text{e2}, \rho \oplus \{\text{id} := \ell\}, \sigma_2 \oplus \{\ell := w\} \Downarrow v, \sigma_3}{\text{Let}(\text{id}, \text{e1}, \text{e2}), \rho, \sigma_1 \Downarrow v, \sigma_3} \text{ELet} \\
\\
\frac{}{\text{Proc}(\text{id}, \text{e}), \rho, \sigma \Downarrow (\text{id}, \text{e}, \rho), \sigma} \text{EProc} \\
\\
\frac{\text{e1}, \rho, \sigma \Downarrow (\text{id}, \text{e}, \tau), \sigma_1 \quad \text{e2}, \rho, \sigma_1 \Downarrow w, \sigma_2 \quad \ell \notin \text{dom}(\sigma_2) \quad \text{e}, \tau \oplus \{\text{id} := \ell\}, \sigma_2 \oplus \{\ell := w\} \Downarrow v, \sigma_3}{\text{App}(\text{e1}, \text{e2}), \rho, \sigma \Downarrow v, \sigma_3} \text{EApp} \\
\\
\frac{\text{e}, \rho, \sigma \Downarrow v, \sigma'}{\text{Set}(\text{id}, \text{e}), \rho, \sigma \Downarrow \text{unit}, \sigma' \oplus \{\rho(\text{id}) := v\}} \text{ESet} \\
\\
\frac{\rho(\text{id}) \notin \mathbb{L} \text{ or } \rho(\text{id}) \notin \text{dom}(\sigma)}{\text{Set}(\text{id}, \text{e}), \rho, \sigma \Downarrow \text{error}, \sigma} \text{ESetErr} \\
\\
\frac{n > 0 \quad (\text{ei}, \rho, \sigma_i \Downarrow v_i, \sigma_{i+1})_{i \in 1..n}}{\text{BeginEnd}([\text{e1}; \dots; \text{en}]), \rho, \sigma_1 \Downarrow v_n, \sigma_{n+1}} \text{EBeginEndNE} \\
\\
\frac{}{\text{BeginEnd}([]), \rho, \sigma \Downarrow \text{unit}, \sigma} \text{EBeginEndE}
\end{array}$$

Figure 3.2: Evaluation rules for IMPLICIT-REFS (error propagation rules are omitted)

```

5  ...
6  | Let(v,def,body) ->
7    eval_expr def >>= fun ev ->      (* evaluate definition *)
8    let l = Store.new_ref g_store ev  (* allocate it in the store *)
9    in extend_env v (RefVal l) >>+    (* extend env with new key-value pair *)
10   eval_expr body                    (* eval body in extended env *)
11 | Set(id,e) ->
12   eval_expr e >>= fun ev ->      (* eval RHS *)
13   apply_env id >>=
14   int_of_refVal >>= fun l ->      (* make sure id is mapped to location *)
15   Store.set_ref g_store l ev >>= fun _ -> (* update the store *)
16   return UnitVal
17 | BeginEnd([]) ->
18   return UnitVal
19 | BeginEnd(es) ->
20   sequence (List.map eval_expr es) >>= fun vs ->
21   return (List.hd (List.rev vs))
22 | Debug(_e) ->
23   string_of_env >>= fun str_env ->
24   let str_store = Store.string_of_store string_of_expval g_store
25   in (print_endline (str_env^"\n"^str_store);
26      error "Debug called")
27 | _ -> failwith ("Not implemented: "^string_of_expr e)

```

3.3.3.3 letrec Revisited

Our implementation of `letrec` in REC consisted in adding a specific entry in the environment to signal the declaration of a recursive function. Then, upon lookup, a closure was created on the fly. This is the code from REC. The highlighted excerpt `Ok (ProcVal (par,body,env))` indicates that a closure is being created.

```

type env =
2 | EmptyEnv
3 | ExtendEnv of string*exp_val*env
4 | ExtendEnvRec of string*string*expr*env

6 let rec apply_env : string -> exp_val ea_result =
7   fun id env ->
8     match env with
9     | EmptyEnv -> Error (id^" not found!")
10    | ExtendEnv(v,ev,tail) ->
11      if id=v
12      then Ok ev
13      else apply_env id tail
14    | ExtendEnvRec(v,par,body,tail) ->
15      if id=v
16      then Ok (ProcVal (par,body,env))
17      else apply_env id tail

```

ds.ml

We could follow the same approach in IMPLICIT-REFS. But there is a better way, which avoids having to create closures on the fly. The idea is to allow circular environments. That is, an environment `env` that has an entry to a location on the store that holds a closure whose environment has a reference to this same location.

So we first remove the special entry in environments for `letrec` declarations since they will no longer be needed:

```

type env =
2   | EmptyEnv
  | ExtendEnv of string*exp_val*env
4   | ExtendEnvRec of string*string*expr*env

6   let rec apply_env : string -> exp_val ea_result =
    fun id env ->
8     match env with
    | EmptyEnv -> Error (id^" not found!")
10    | ExtendEnv(v, ev, tail) ->
        if id=v
12        then Ok ev
        else apply_env id tail
14    | ExtendEnvRec(v, par, body, tail) ->
        if id=v
16        then Ok (ProcVal (par, body, env))
        else apply_env id tail

```

ds.ml

We now use “back-patching” to code the circular environment:

```

1   let rec eval_expr : expr -> exp_val ea_result =
    fun e ->
3     match e with
    | Letrec([(id, par, _, _, e)], target) ->
5       let l = Store.new_ref g_store UnitVal in
        extend_env id (RefVal l) >>+
7       (lookup_env >>= fun env ->
        Store.set_ref g_store l (ProcVal(par, e, env)) >>= fun _ ->
9       eval_expr target
    )

```

interp.ml



Paranthesis right after ($\gg+$) are necessary since ($\gg=$) and ($\gg+$) are left-associative. Remove them, execute the resulting interpreter on an example expression and explain what goes wrong.

3.4 Parameter Passing Methods

We consider several parameter passing methods in IMPLICIT-REFS.

3.4.1 Call-by-Value

This method consists in first evaluating the argument, before passing on its value to the function. This is the parameter passing method we have implemented in PROC and all the languages that extend it.

3.4.2 Call-by-Reference

If the argument to a function is a variable, then we provide a copy of its reference to the function. Otherwise, the argument is processed just like in call-by-value. For example, evaluation of

```

1 let x = 2
2 in let f = proc (z) { set z = z+1 }
3 in begin
4   (f x);
5   x
6 end

```

returns `Ok (NumVal 2)` in IMPLICIT-REFS. However, using call-by-reference, it will return `Ok (NumVal 3)`. It is helpful to place a breakpoint inside the body of the `f`, evaluate the resulting expression and examine the environment and store.

3.4.2.1 Modifying the Interpreter

```

1 let rec value_of_operand : expr -> exp_val ea_result =
2   fun e ->
3     match e with
4     | Var(id) -> apply_env id
5     | _ -> eval_expr e >>= fun ev ->
6       return (RefVal (Store.new_ref g_store ev))
7   and
8     apply_clos =
9       ...
10  and
11    eval_expr : expr -> exp_val ea_result =
12    fun e ->
13      match e with
14      | ...
15      | App(e1,e2) ->
16        eval_expr e1 >>=
17        clos_of_procVal >>= fun clos ->
18        eval_expr e2 >>=
19        value_of_operand e2 >>=
20        apply_clos clos

```

interp.ml

The following example shows how one may swap the contents of two variables:

```

1 let a = 2
2 in let b = 1
3 in let f = proc (x) { proc (y) {
4   let temp = x
5   in begin
6     set x = y;
7     set y = temp
8   end }}
9 in begin
10   ((f a) b);
11   a
12 end

```

Returns `Ok (NumVal 1)`.

3.4.3 Call-by-Name

Consider the following expression. What is the result of its evaluation?

```

letrec infinite_loop (x) = (infinite_loop (x+1))
2 in let f = proc (y) { 11 }
in (f (infinite_loop 0))

```

The parameter z is not used. However, the argument to f , namely $(\text{infinite_loop } 0)$, is evaluated all the same. The call-by-name parameter passing method consists in freezing the evaluation of arguments until they are actually needed. This is achieved as follows. In an application $\text{App}(e_1, e_2)$, if e_2 is an identifier, then call-by-name proceeds just like call-by-reference: it passes a copy of the address of the identifier, after looking it up in the environment, as an argument to the parameter. Regardless of whether the value of e_2 will be used, copying an address is a constant time operation, hence not costly. However, if e_2 is an expression different from an identifier, then its evaluation does not take place. Rather, e_2 together with the current environment are stored for later evaluation. The pair consisting of e_2 and the current environment is called a thunk. A thunk is just a closure without the formal parameter. We next implement call-by-name, the parameter passing method that implements this idea. We will use the code for call-by-reference as a starting point.

We begin by adding thunks to the set of expressed values:

```

type exp_val =
2 | NumVal of int
  | BoolVal of bool
4 | ProcVal of string*expr*env
  | UnitVal
6 | Thunk of expr*env

```

ds.ml

Note that we use `Thunk` rather than `ThunkVal` as the name of the constructor, to emphasize that a thunk is not a run-time value that may be returned as the result of a evaluation. Next we update `value_of_operand`. If the argument or operand is a variable, our interpreter behaves just like in call-by-reference. However, if it is not a variable, we create a thunk:

```

let rec value_of_operand =
2 fun op ->
  match op with
4 | Var id -> apply_env id
  | _ ->
    lookup_env >>= fun en ->
      return (RefVal (Store.new_ref g_store (Thunk(op, en))))

```

interp.ml

Finally, we have to consider what happens when we lookup the contents of the store through the reference assigned to a variable in the environment and it consists of a thunk `Thunk(e, en)`. We “thaw” the thunk by evaluating e under the environment en :

```

1 let rec eval_expr : expr -> exp_val ea_result =
  fun e ->
3 match e with
  | Int(n) -> return (NumVal n)
5 | Var(id) ->
    apply_env id >>=

```

```

7   int_of_refVal >>=
    Store.deref g_store >>= fun ev ->
9   (match ev with
    | Thunk(e,en) -> return en >>+ eval_expr e
11  | _ -> return ev)
    ...

```

interp.ml

Evaluate the example from the beginning of this section in CBN. Then do the same but this time inserting a breakpoint in the body of `f`:

```

letrec infinite_loop (x) = (infinite_loop (x+1))
2 in let f = proc (y) { debug(11) }
   in (f (infinite_loop 0))

```

3.4.4 Call-by-Need

One drawback of call-by-name is that a thunk is “thawed” every time it is needed. Consider the following example:

```

1 letrec f(x) = if zero?(x) then 1 else x*(f (x-1))
  in let g = proc (y) { y+y+y+y }
3  in (g (f 5))

```

Here the factorial of 5 is computed four times, one for each occurrence of `y` in `g`. A more reasonable approach is to do this once, the first time, and then store the result for further uses. This optimization technique is called memoization. Call-by-need consists in applying this optimization technique to call-by-name. All we need to do is update the implementation for the variable constructor in `eval_expr`:

```

1 let rec eval_expr : expr -> exp_val ea_result =
  fun e ->
3   match e with
  | Int(n) -> return @@ NumVal n
  | Var(id) ->
5     apply_env id >>=
7     int_of_refVal >>= fun l ->
      Store.deref g_store l >>= fun ev ->
9     (match ev with
     | Thunk(e,en) ->
11      return en >>+
        eval_expr e >>= fun ev ->
13      Store.set_ref g_store l ev >>= fun _ ->
        return ev
     | _ -> return ev)
    ...

```

interp.ml

The factorial example above now involves computing factorial of 5 just once.

Call-by-need and call-by-name may return different results for effectful computation. Evaluate the following expression in CBN and CBNeed:

```

let f = let count = 0
2   in proc (d) {
      begin

```



```

4         set count = count + 1;
          count
6     end }
in let g = proc (x) { x+x }
8 in (g (f 0))

```

3.5 Exercises

Exercise 3.5.1. Depict the environment and store at the breakpoint for the following EXPLICIT-REFS programs:

1.

```
let a = 3
in let b = newref(if zero?(a) then 1 else 2)
in debug(a)
```

2.

```
let a = newref(newref(2+1))
in debug(a)
```

3.

```
let a = (let s = newref(0)
in proc(i) {
begin
setref(s, deref(s)+i);
deref(s)
end
})
in debug(a)
```

Exercise 3.5.2. Consider the following extension of LET with records (Exercise 2.2.6). It has the same syntax except that one can declare a field to be mutable by using `<=` instead of `=`. For example, the `ssn` field is immutable but the `age` field is mutable; `age` is then updated to 31:

```
let p = {ssn = 10; age <= 30}
in begin
p.age <= 31;
p.age
end
```

Evaluating this expression should produce `Ok (NumVal 31)`. This other expression should produce `Ok (RecordVal [{"ssn", (false, NumVal 10)}; {"age", (true, RefVal 1)}])`:

```
let p = {ssn = 10; age <= 30}
in begin
p.age <= 31;
p
end
```

Updating an immutable field should not be allowed. For example, the following expression should report an error `Error "Field not mutable"`:

```

let p = { ssn = 10; age = 20}
in begin
  p.age <= 21;
  p.age
end

```

The abstract syntax requires modifying the *Record* constructor and adding a new one for field update:

```

type expr =
  ...
  | Record of (string*(bool*expr)) list
  | SetField of expr*string*expr

```

For example,

```

# parse "
let p = {ssn = 10; age <= 30}
in begin
  p.age <= 31;
  p
end";;
- : expr =
Let ("p", Record [("ssn", (false, Int 10)); ("age", (true, Int 30))],
  BeginEnd [SetField (Var "p", "age", Int 31); Var "p"])

```

utop

Here *false* indicates the field is immutable and *true* that it is mutable. You are asked to implement the interpreter extension. The *RecordVal* constructor has been updated for you.

```

type exp_val =
  | NumVal of int
  | BoolVal of bool
  | ProcVal of string*expr*env
  | PairVal of exp_val*exp_val
  | TupleVal of exp_val list
  | RecordVal of (string*(bool*exp_val)) list

```

ds.ml

As for *eval_expr*, the case for *Record* has already been updated for you. You are asked to update *Proj* and complete *SetField*:

```

let rec eval_expr : expr -> exp_val ea_result = fun e ->
  match e with
  | Record(fs) ->
    sequence (List.map process_field fs) >>= fun evs ->
    return (RecordVal (addIds fs evs))
  | Proj(e, id) ->
    error "update"
  | SetField(e1, id, e2) ->
    error "implement"
and
process_field (_id, (is_mutable, e)) =
  eval_expr e >>= fun ev ->
  if is_mutable
  then return (RefVal (Store.new_ref g_store ev))
  else return ev

```

Exercise 3.5.3. *Depict the environment and store extant at the breakpoint in the following IMPLICIT-REFS expressions.*

1.

```
let a = 2
in let b = 3
in begin
  set a = b;
  debug(a)
end
```
2.

```
let a = 2
in let b = a
in begin
  set b = 3;
  debug(a)
end
```
3.

```
let a = 2
in let b = proc(x) {
  begin
    set a = x;
    debug(a)
  end
}
in (b 3)
```
4.

```
let a = 2
in let b = proc(x) {
  begin
    set a = x;
    a
  end
}
in (b 3) + debug((b 4))
```

Exercise 3.5.4. *Depict the environment and store at the breakpoint first assuming call-by-reference as parameter passing method and then call-by-value.*

1.

```
let a=1
in let b=2
in let f = proc (x) { proc (d) { debug(set x=x+d) }}
in ((f a) (b+1))
```


Chapter 4

Types

This chapter extends the REC language to support type-checking.

4.1 CHECKED

4.1.1 Concrete Syntax

```
⟨Expression⟩ ::= ⟨Number⟩
⟨Expression⟩ ::= ⟨Identifier⟩
⟨Expression⟩ ::= ⟨Expression⟩⟨BOP⟩⟨Expression⟩
⟨Expression⟩ ::= zero?(⟨Expression⟩)
⟨Expression⟩ ::= if ⟨Expression⟩ then ⟨Expression⟩ else ⟨Expression⟩
⟨Expression⟩ ::= let ⟨Identifier⟩ = ⟨Expression⟩ in ⟨Expression⟩
⟨Expression⟩ ::= (⟨Expression⟩)
⟨Expression⟩ ::= proc(⟨Identifier⟩:⟨Type⟩){⟨Expression⟩}
⟨Expression⟩ ::= (⟨Expression⟩⟨Expression⟩)
⟨Expression⟩ ::= letrec{⟨Identifier⟩(⟨Identifier⟩:⟨Type⟩):⟨Type⟩ = ⟨Expression⟩}+ in ⟨Expression⟩

⟨BOP⟩ ::= + | - | * | /

⟨Type⟩ ::= int
⟨Type⟩ ::= bool
⟨Type⟩ ::= ⟨Type⟩->⟨Type⟩
⟨Type⟩ ::= (⟨Type⟩)
```

4.1.2 Abstract Syntax

```
type expr =
2 | Var of string
  | Int of int
4 | Sub of expr*expr
  | Let of string*expr*expr
```

```

6 | IsZero of expr
  | ITE of expr*expr*expr
8 | Proc of string*texpr option*expr
  | App of expr*expr
10 | Letrec of rdec*s*expr
and
12 rdec*s = (string*string*texpr option*texpr option*expr) list
and
14 texpr =
  | IntType
16 | BoolType
  | FuncType of texpr*texpr

```

4.1.3 Type-Checker

We specify the behavior of our type-checker before implementing it, much like we do with interpreters. For this task we use **type systems**. A type system is an inductive set that helps identify a subset of the expressions that are considered to be well-typed. The elements of the inductive set are called **type judgements**. Which type judgements belong to the set and which don't is determined by a set of type rules. A type judgement is an expression of the form

$$\Gamma \vdash e : t$$

where Γ is a type environment, e is an expression in CHECKED, and t is a type expression. These components together with the type rules are introduced below.

4.1.3.1 Specification

As mentioned, a type judgement consists of a of type environment, an expression in CHECKED and a type. Types are defined as follows:

$$t ::= \text{int} \mid \text{bool} \mid t \rightarrow t$$

A **type context** is a partial function that assigns a type to an identifier. Type contexts are required for typing expressions that contain free variables. For example, an expression such as $x+2$ will require that we have the type of x at our disposal in order to determine whether $x+2$ is typable at all. If the type of x were `bool`, then it is not typable; but if the type of x is `int`, then it is. Type contexts are defined as follows:

$$\Gamma ::= \epsilon \mid \Gamma, id : t$$

We use ϵ to denote the empty type environment. Also, $\Gamma, id : t$ assigns type t to identifier id and behaves as Γ for identifiers different from id . We assume that Γ does not have repeated entries for the same identifier. An example of a type contexts is $\epsilon, x : \text{int}, y : \text{bool}$. We abbreviate it as $x : \text{int}, y : \text{bool}$.

The type rules are given in Figure 4.1.

$$\begin{array}{c}
\frac{}{\Gamma \vdash n : \text{int}} \text{TInt} \quad \frac{\Gamma(x) = t}{\Gamma \vdash x : t} \text{TVar} \quad \frac{\Gamma \vdash e : \text{int}}{\Gamma \vdash \text{zero?}(e) : \text{bool}} \text{TIsZero} \\
\\
\frac{\Gamma \vdash e1 : \text{int} \quad \Gamma \vdash e2 : \text{int}}{\Gamma \vdash e1 - e2 : \text{int}} \text{TSub} \\
\\
\frac{\Gamma \vdash e1 : \text{bool} \quad \Gamma \vdash e2 : t \quad \Gamma \vdash e3 : t}{\Gamma \vdash \text{if } e1 \text{ then } e2 \text{ else } e3 : t} \text{TITE} \\
\\
\frac{\Gamma \vdash e1 : t1 \quad \Gamma, id : t1 \vdash e2 : t2}{\Gamma \vdash \text{let } id = e1 \text{ in } e2 : t2} \text{TLet} \\
\\
\frac{\Gamma \vdash \text{rator} : t1 \rightarrow t2 \quad \Gamma \vdash \text{rand} : t1}{\Gamma \vdash (\text{rator } \text{rand}) : t2} \text{TApp} \\
\\
\frac{\Gamma, id : t1 \vdash e : t2}{\Gamma \vdash \text{proc } (id : t1) \{e\} : t1 \rightarrow t2} \text{TProc}
\end{array}$$

Figure 4.1: Type Rules for CHECKED

4.1.3.2 Towards and Implementation

Our type checker will behave very much like our interpreter, except that instead of manipulating runtime values such as integers and booleans, it manipulates types like `int` and `bool`. One might say that a type checker is a symbolic evaluator, where our symbolic values are the types. This analogy allows us to apply the ideas we have developed on well-structuring an evaluator to our type checker. Thus one might be tempted to state the type of our type-checker as

```
chk_expr : expr -> texpr ea_result
```

reflecting that, given an expression, it returns a function that given a type environment returns either a type or an error. Note, however, that `ea_result` abstracts environments, and not type environments:

```
type 'a ea_result = env -> 'a result
```

We could create a new type constructor, let us call it `tea_result`, where `env` is replaced with `tenv`:

```
type 'a tea_result = tenv -> 'a result
```

But we would also have to duplicate all of `return`, `error`, `(>>=)`, `lookup_env`, etc. to support this new type and end up having two copies of all these operations (one supporting `ea_result` and one supporting `tea_result`) with the exact same code. Since the only difference between `ea_result` and `tea_result` is the kind of environment they abstract over, we choose to define a more general type constructor `a_result` (“a” for “abstracted”) and have both of these be instances of them:

```
type ('a,'b) a_result = 'b -> 'a result
```

```

1 type 'a result = Ok of 'a | Error of string
2
3 type ('a,'b) a_result = 'b -> 'a result
4
5 let return : 'a -> ('a,'b) a_result =
6   fun v ->
7     fun env -> Ok v
8
9 let error : string -> ('a,'b) a_result =
10  fun s ->
11    fun env -> Error s
12
13 let (>=>) : ('a,'c) a_result -> ('a -> ('b,'c) a_result) -> ('b,'c) a_result =
14  fun c f ->
15    fun env ->
16      match c env with
17      | Error err -> Error err
18      | Ok v -> f v env
19
20 let (>>+) : ('b,'b) a_result -> ('a,'b) a_result -> ('a,'b) a_result =
21  fun c d ->
22    fun env ->
23      match c env with
24      | Error err -> Error err
25      | Ok newenv -> d newenv

```

reM.ml

Figure 4.2: The `a_result` type

Notice that, contrary to `ea_result` and `tea_result`, the type constructor `a_result` is parameterized over two types,

1. the type `a` representing the result of the computation, and
2. the type `b` representing that over which the function type is being abstracted over.

The type `('a,'b) a_result`, together with its supporting operations are declared in Figure 4.2. Notice that the code for the supporting operations, namely `return`, `error`, `(>=>)` and `(>>+)` is exactly the same as before. The only difference is their type. That being said, we still call the formal parameter of type `'b` in these operations, `env`.

With the newly declared type constructor `a_result` in place, we can now redefine `ea_result` and `tea_result` as instances of it. Indeed, `ea_result` is simply defined as:

```
type 'a ea_result = ('a,env) a_result
```

and `tea_result` is defined as:

```
type 'a tea_result = ('a,tenv) a_result
```

4.1.3.3 Implementation

We next address the implementation of the type-checker for `CHECKED`. The code is given in Figure 4.3.


```

let rec chk_expr : expr -> texpr tea_result =
2   fun e ->
   match e with
4   | Int _n -> return IntType
   | Var id -> apply_tenv id
6   | IsZero(e) ->
      chk_expr e >>= fun t ->
8     if t=IntType
      then return BoolType
      else error "isZero: expected argument of type int"
   | Add(e1,e2) | Sub(e1,e2) | Mul(e1,e2) | Div(e1,e2) ->
      chk_expr e1 >>= fun t1 ->
12     chk_expr e2 >>= fun t2 ->
      if (t1=IntType && t2=IntType)
      then return IntType
      else error "arith: arguments must be ints"
   | ITE(e1,e2,e3) ->
      chk_expr e1 >>= fun t1 ->
18     chk_expr e2 >>= fun t2 ->
      chk_expr e3 >>= fun t3 ->
      if (t1=BoolType && t2=t3)
      then return t2
      else error "ITE: condition not bool/types of then-else do not match"
24   | Let(id,e,body) ->
      chk_expr e >>= fun t ->
      extend_tenv id t >>+
      chk_expr body
28   | Proc(var,t1,e) ->
      extend_tenv var t1 >>+
      chk_expr e >>= fun t2 ->
30     return (FuncType(t1,t2))
   | App(e1,e2) ->
      chk_expr e1 >>=
34     pair_of_funcType "app: " >>= fun (t1,t2) ->
      chk_expr e2 >>= fun t3 ->
36     if t1=t3
      then return t2
      else error "app: type of argument incorrect"
   | Debug(_e) ->
40     string_of_tenv >>= fun str ->
      print_endline str;
42     error "Debug: reached breakpoint"
   | _ -> failwith "chk_expr: implement"

```

checker.ml

Figure 4.3: Type checker for CHECKED

```

1 # chk "
let add = proc (x:int) { proc (y:int) { x+y }} in (add 1)";;
3 - : texpr ReM.result = Ok (FuncType (IntType, IntType))

```

utop

4.1.4 Adding Letrec

The typing rule for letrec is as follows:

$$\frac{\begin{array}{c} \Gamma, \text{par} : tVar, \text{id} : tPar \rightarrow tRes \vdash \text{body} : tRes \\ \Gamma, \text{id} : tPar \rightarrow tRes \vdash \text{target} : t \end{array}}{\Gamma \vdash \text{letrec id (par:tPar):tRes} = \text{body in target} : t} \text{TRec}$$

```

1 let rec chk_expr : expr -> texpr tea_result = function
  | Letrec([(id,param,tPar,tRes,body)],target) ->
3   extend_tenv id (FuncType(tPar,tRes)) >>+
   (extend_tenv param tPar >>+
5    chk_expr body >>= fun t ->
   if t=tRes
7   then chk_expr target
   else error "LetRec: Type of rec. function does not match declaration")

```

4.1.5 Exercises

Exercise 4.1.1. Provide typing derivations for the following expressions:

1. `if zero?(8) then 1 else 2`
2. `if zero?(8) then zero?(0) else zero?(1)`
3. `proc (x:int) { x-2 }`
4. `proc (x:int) { proc (y:bool) { if y then x else x-1 } }`
5. `let x=3 in let y = 4 in x-y`
6. `let two? = proc(x:int) { if zero?(x-2) then 0 else 1 } in (two? 3)`

Exercise 4.1.2. Recall that an expression e is typable, if there exists a type environment Γ and a type expression t such that the typing judgement $\Gamma \vdash e : t$ is derivable. Argue that the expression $x \ x$ (a variable applied to itself) is not typable.

Exercise 4.1.3. Give a typable term of each of the following types, justifying your result by showing a type derivation for that term.

1. `bool->int`
2. `(bool -> int) -> int`
3. `bool -> (bool -> bool)`

4. $(s \rightarrow t) \rightarrow (s \rightarrow t)$, for any types s and t .

Exercise 4.1.4. Show that the following term is typable:

```
letrec double (x:int):int = if zero?(x)
                           then 0
                           else (double (x-1)) + 2
in double
```

Exercise 4.1.5. What is the result of evaluating the following expressions in CHECKED?

```
utop # chk "
letrec double (x:int):int = if zero?(x)
                           then 0
                           else (double (x-1)) + 2
in (double 5)";;
```

utop

```
utop # chk "
letrec double (x:int):int = if zero?(x)
                           then 0
                           else (double (x-1)) + 2
in double";;
```

utop

```
utop # chk "
letrec double (x:int):bool = if zero?(x)
                             then 0
                             else (double (x-1)) + 2
in double";;
```

utop

```
utop # chk "
letrec double (x:int):bool = if zero?(x)
                             then 0
                             else 1
in double";;
```

utop

Exercise 4.1.6. Consider the extension of Exercise 2.2.4 where pairs are added to our language. In order to extend type-checking to pairs we first add pair types to the concrete syntax of types:

$$\begin{aligned} \langle \text{Type} \rangle &::= \text{int} \\ \langle \text{Type} \rangle &::= \text{bool} \\ \langle \text{Type} \rangle &::= \langle \text{Type} \rangle \rightarrow \langle \text{Type} \rangle \\ \langle \text{Type} \rangle &::= \langle \text{Type} \rangle * \langle \text{Type} \rangle \\ \langle \text{Type} \rangle &::= (\langle \text{Type} \rangle) \end{aligned}$$

Recall from Exercise 2.2.4 that expressions are extended with a $\text{pair}(e_1, e_2)$ construct to build new pairs and an $\text{unpair } (x, y) = e_1 \text{ in } e_2$ construct that given an expression e_1 that evaluates to a pair, binds x and y to the first and second component of the pair, respectively, in e_2 . Here are some examples of expressions in the extended language:

```

pair(3,4)

pair(pair(3,4),5)

pair(zero?(0),3)

pair(proc (x:int) { x-2 },4)

proc (z:int*int) { unpair (x,y)=z in x }

proc (z:int*bool) { unpair (x,y)=z in pair(y,x) }

```

You are asked to give typing rules for each of the two new constructs.

Exercise 4.1.7. Consider the following extension of CHECKED with records, as introduced in Exercise 2.2.6. The concrete syntax for the new type constructor for records is given by the second to last production below:

```

<Type> ::= int
<Type> ::= bool
<Type> ::= <Type> -> <Type>
<Type> ::= { {<Identifier> = <Type> }+(;) }
<Type> ::= ( <Type> )

```

The abstract syntax is as follows:

```

type expr =
  | Var of string
  | Int of int
  | Sub of expr*expr
  | Let of string*expr*expr
  | IsZero of expr
  | ITE of expr*expr*expr
  | Proc of string*teexpr*expr
  | App of expr*expr
  | Letrec of rdecs*expr
  | Record of (string*expr) list
  | Proj of expr*string
and
  rdecs = (string*string*teexpr option*teexpr option*expr) list
and
  teexpr =
    | IntType
    | BoolType
    | FuncType of teexpr*teexpr
    | RecordType of (string*teexpr) list

```

For example,

1. {age=2; height=3} should have type {age:int; height:int}.
2. {age=2; present=zero?(0)} should have type {age:int; present:bool}.
3. {inc = proc(x:int) {x+1}; dec = proc(x:int) {x-1}} should have type {inc:int->int; dec:int->int}.

4. $\{inc = proc(x:int) \{x+1\}; dec = proc(x:int) \{x-1\}\}.inc$ should have type $int \rightarrow int$.
5. $\{\}$ should produce a type error since empty records are not allowed.
6. $\{age=2; height=3\}.weight$ should produce a type error since there is no field named *weight*.

The additional typing rules are:

$$\frac{\Gamma \vdash e_1 : t_1 \quad \dots \quad \Gamma \vdash e_n : t_n \quad n > 0 \quad l_i, i \in 1..n, \text{distinct}}{\Gamma \vdash \{ l_1=e_1; \dots; l_n=e_n \} : \{l_1 : t_1; \dots; l_n : t_n\}} \text{TRec}$$

$$\frac{\Gamma \vdash e : \{l_1 : t_1; \dots; l_n : t_n\} \quad l = l_i, \text{ for some } i \in 1..n}{\Gamma \vdash e.l : t_i} \text{TProj}$$

Extend the type checker *chk_expr* to deal with the two new constructs.

Chapter 5

Simple Object-Oriented Language

5.1 SOOL

5.1.1 Concrete Syntax

The concrete syntax for SOOL is presented below. We'll introduce it in stages.

$$\langle \text{Program} \rangle ::= \langle \text{ClassDecl} \rangle^* \langle \text{Expression} \rangle$$

A SOOL program is a (possibly empty) list of class declarations followed by a main expression.

$$\begin{aligned} \langle \text{Expression} \rangle &::= \text{new } \langle \text{Identifier} \rangle (\langle \text{Expression} \rangle^{*(\cdot)}) \\ \langle \text{Expression} \rangle &::= \text{self} \\ \langle \text{Expression} \rangle &::= \text{send } \langle \text{Expression} \rangle \langle \text{Identifier} \rangle (\langle \text{Expression} \rangle^{*(\cdot)}) \\ \langle \text{Expression} \rangle &::= \text{super } \langle \text{Identifier} \rangle (\langle \text{Expression} \rangle^{*(\cdot)}) \\ \langle \text{Expression} \rangle &::= \text{list} (\langle \text{Expression} \rangle^{*(\cdot)}) \\ \langle \text{Expression} \rangle &::= \text{hd} (\langle \text{Expression} \rangle) \\ \langle \text{Expression} \rangle &::= \text{tl} (\langle \text{Expression} \rangle) \\ \langle \text{Expression} \rangle &::= \text{empty?} (\langle \text{Expression} \rangle) \\ \langle \text{Expression} \rangle &::= \text{cons} (\langle \text{Expression} \rangle, \langle \text{Expression} \rangle) \end{aligned}$$

An expression can be any expression in IMPLICIT-REFS together with four new object-oriented specific expressions (the first four listed above) and some additional ones that support lists. Only the new productions that are added to the grammar for IMPLICIT-REFS are depicted above.

$$\langle \text{ClassDecl} \rangle ::= \langle \text{Identifier} \rangle \text{ extends } \langle \text{Identifier} \rangle \{ \langle \text{FieldDecl} \rangle^* \langle \text{MethodDecl} \rangle^* \}$$

A class declaration consists of the name of the class being defined, the name of the superclass, a list of field declarations and a list of method declarations.

$$\begin{aligned} \langle \text{FieldDecl} \rangle &::= \text{field } \langle \text{Identifier} \rangle \\ \langle \text{MethodDecl} \rangle &::= \text{method } \langle \text{Identifier} \rangle (\langle \text{Identifier} \rangle^{*(\cdot)}) \{ \langle \text{Expression} \rangle \} \end{aligned}$$

A field declaration is just the keyword `field` followed by an identifier, the name of the field. A method declaration consists of the keyword `method`, an identifier representing the name of the method, a list of formal parameters between parenthesis, and the body.

5.1.2 Abstract Syntax

```

type
2   prog = AProg of (cdecl list)*expr
and
4   expr =
    | Var of string
6   | Int of int
    | Add of expr*expr
8   | Sub of expr*expr
    | Mul of expr*expr
10  | Div of expr*expr
    | Abs of expr
12  | Let of string*expr*expr
    | IsZero of expr
14  | ITE of expr*expr*expr
    | Proc of string*texpr option*expr
16  | App of expr*expr
    | Letrec of string*string*texpr option*texpr option*expr*expr
18  | Set of string*expr
    | BeginEnd of expr list
20  | Self
    | Send of expr*string*expr list
22  | Super of string*expr list
    | NewObject of string*expr list
24  | Cons of expr*expr
    | Hd of expr
26  | Tl of expr
    | IsEmpty of expr
28  | List of expr list
    | Debug of expr
30 and
    cdecl = Class of string*string*string option*(string*texpr option) list*mdecl list
32 and
    mdecl = Method of string*texpr option*(string*texpr option) list*expr

```

We omit the type declaration for `texpr`; this will be given later.

5.1.3 Interpreter

5.1.3.1 Specification

5.1.3.2 Implementation

Chapter 6

Modules

6.1 Syntax

SIMPLE-MODULES is an extension to the EXPLICIT-REFS language. A program in SIMPLE-MODULES consists of a list of module declarations together with an expression (the “main” expression). Here is an example that consists of one module declaration, the module called `m1`, and a main expression consisting of a `let` expression. A module has an interface and a body.

```
1 module m1
2   interface
3     [a : int
4       b : int
5       c : int]
6   body
7     [a = 33
8       x = a-1 (* =32 *)
9       b = a-x (* = 1 *)
10      c = x-b] (* =31 *)
11 let a = 10
12 in ((from m1 take a) - (from m1 take b))-a
```

6.1.1 Concrete Syntax

A program in SIMPLE-MODULES consists of a possible empty sequence of module declarations followed by an expression:

$$\langle \text{Program} \rangle ::= \{ \langle \text{ModuleDefn} \rangle \}^* \langle \text{Expression} \rangle$$

Expressions are the those of REC but with an extra production that we refer to as a qualified variable reference:

$$\langle \text{Expression} \rangle ::= \text{from } \langle \text{Identifier} \rangle \text{ take } \langle \text{Identifier} \rangle$$

The concrete syntax of modules is given by the following grammar:

$$\begin{aligned}
\langle \text{ModuleDefn} \rangle &::= \text{module } \langle \text{Identifier} \rangle \text{ interface } \langle \text{Iface} \rangle \text{ body } \langle \text{ModuleBody} \rangle \\
\langle \text{Iface} \rangle &::= [\{ \langle \text{Decl} \rangle \}^*] \\
\langle \text{Decl} \rangle &::= \langle \text{Identifier} \rangle : \langle \text{Type} \rangle \\
\langle \text{ModuleBody} \rangle &::= [\{ \langle \text{Defn} \rangle \}^*] \\
\langle \text{Defn} \rangle &::= \langle \text{Identifier} \rangle = \langle \text{Expression} \rangle
\end{aligned}$$

6.1.2 Abstract Syntax

```

type expr =
2   ...
  | QualVar of string*string
4 and
  texpr =
6   | IntType
  | BoolType
8   | UnitType
  | FuncType of texpr*texpr
10  | RefType of texpr
and
12  vdecl = string*texpr
and
14  vdef = string*expr

16 type interface = ASimpleInterface of vdecl list
type module_body = AModBody of vdef list
18 type module_decl = AModDecl of string*interface*module_body
type prog = AProg of (module_decl list)*expr

```

6.2 Interpreter

6.2.1 Specification

The set of results is the same as that for EXPLICIT-REFS. In particular, the set of expressed values consists of integers, booleans, unit, closures and locations:

$$\text{EV} := \mathbb{Z} \cup \mathbb{B} \cup \text{U} \cup \text{CL} \cup \text{L}$$

There are three kinds of evaluation judgements in SIMPLE-MODULES, one for programs, one for expressions, and a third auxiliary one used to evaluate module definitions. The evaluation judgement for programs is:

$$\text{AProg}(\text{mdecls}, e), \rho, \sigma \Downarrow r, \sigma'$$

A program $\text{AProg}(\text{mdecls}, e)$ consists of a sequence of module declarations mdecl and a main expression e . Also, ρ is the initial environment and σ the initial store. The result of the evaluation is r and the updated store is σ' . Evaluation judgements for expressions are similar to those of EXPLICIT-REFS:

$$e, \rho, \sigma \Downarrow r, \sigma'$$

$$\begin{array}{c}
\frac{\text{mdecls}, \rho, \sigma \Downarrow \rho', \sigma' \quad \text{e}, \rho', \sigma' \Downarrow r, \sigma''}{\text{AProg}(\text{mdecls}, \text{e}), \rho, \sigma \Downarrow r, \sigma''} \text{EProg} \\
\\
\frac{}{\epsilon, \rho, \sigma \Downarrow \rho, \sigma} \text{EMDeclsEmpty} \\
\\
\frac{\text{body}, \rho, \sigma \Downarrow \rho', \sigma' \quad \text{ms}, \rho \oplus \{\text{id} := \rho'\}, \sigma' \Downarrow \rho'', \sigma''}{\text{AModDecl}(\text{id}, \text{iface}, \text{body}) \text{ ms}, \rho, \sigma \Downarrow \rho'', \sigma'} \text{EMDeclsCons} \\
\\
\frac{}{\epsilon, \rho, \sigma \Downarrow \rho, \sigma} \text{EBValsEmpty} \\
\\
\frac{\text{e}, \rho, \sigma \Downarrow v, \sigma' \quad \text{vs}, \rho \oplus \{\text{id} := v\}, \sigma' \Downarrow \rho', \sigma''}{(\text{id}, \text{e}) \text{ vs}, \rho, \sigma \Downarrow \rho', \sigma''} \text{EBValsCons} \\
\\
\frac{\rho(\text{mid}) = \rho' \quad \rho'(\text{vid}) = v}{\text{QualVar}(\text{mid}, \text{vid}), \rho, \sigma \Downarrow v, \sigma} \text{EQualVar}
\end{array}$$

Figure 6.1: Evaluation Semantics for SIMPLE-MODULES (error propagation rules omitted)

where e is an expression, ρ an environment, σ the initial store, r the result and σ' the final store. The difference is that the environment will also allow for mappings between module identifiers and their bodies. The body of a module will be implemented as an environment too. The third evaluation judgement is:

$$\text{mdecls}, \rho, \sigma \Downarrow \rho', \sigma'$$

Here mdecls is a sequence of module declarations, ρ is an initial environment and σ is an initial store. Evaluation of mdecls will produce an environment ρ' which associates to each module mid in mdecls an environment ρ_{mid} . Evaluation also produces an updated store σ' .

6.2.2 Implementation

We first extend environments to support bindings for modules:

```

type env =
2 | EmptyEnv
  | ExtendEnv of string*exp_val*env
4 | ExtendEnvRec of string*string*expr*env
  | ExtendEnvMod of string*env*env

```

[ds.ml](#)

Evaluation of programs consists in first evaluating all module definitions producing an environment as a result, and then evaluating the main expression using this environment. The former is achieved with the helper function `eval_module_definitions : module_decl list -> env ea_result`. We'll describe this function shortly.

```

1 let eval_prog (AProg(ms,e)) : exp_val ea_result =
2   eval_module_definitions ms >>+
   eval_expr e

```

Evaluation of expressions is just like in EXPLICIT-REFS, except that we must deal with the new case, namely that of a qualified variable `QualVar(module_id,var_id)`.

```

1 let rec eval_expr : expr -> exp_val ea_result =
   fun e ->
3   match e with
   ...
5   | QualVar(module_id,var_id) ->
       apply_env_qual module_id var_id
7   ...

```

The helper function `apply_env_qual` inspects the environment looking for a module name `module_id` and then an identifier `var_id` declared within that module:

```

1 let rec apply_env_qual : string -> string -> exp_val ea_result =
   fun mid id ->
3   fun env ->
       match env with
5   | EmptyEnv -> Error "Key not found"
   | ExtendEnv(key,value,env) -> apply_env_qual mid id env
7   | ExtendEnvRec(key,param,body,env) -> apply_env_qual mid id env
   | ExtendEnvMod(moduleName,bindings,env) ->
9     if mid=moduleName
       then apply_env id bindings
11    else apply_env_qual mid id env

```

Finally, we turn to the above mentioned `eval_module_definitions` helper function. Given a list of module declarations `ms`, it evaluates them one by one using `eval_module_definition` and then returning a value of type `env ea_result` holding the resulting environment.

```

1 let rec eval_expr : expr -> exp_val ea_result =
   ...
3 and
   eval_module_definition : module_body -> env ea_result =
5   fun (AModBody vdefs) ->
       lookup_env >>= fun glo_env -> (* holds all previously declared modules *)
7       List.fold_left
           (fun loc_env (var,decl) ->
13              loc_env >>+
                  (append_env_rev glo_env >>+
                      eval_expr decl >>=
                          extend_env var))
           (empty_env ())
           vdefs
15 and
   eval_module_definitions : module_decl list -> env ea_result =
17   fun ms ->
       List.fold_left
19         (fun curr_en (AModDecl(mname,minterface,mbody)) ->
21             curr_en >>+
                  (eval_module_definition mbody >>=
                      extend_env_mod mname))
           lookup_env
23

```

```

ms
25 and
    eval_prog (AProg(ms,e)) : exp_val ea_result =
27   eval_module_definitions ms >>+
    eval_expr e
29
interp.ml

utop # interp "
2 module m1
  interface
4   [u : int]
  body
6   [u = 44]
module m2
  interface
8   [v : int]
  body
10  [v = (from m1 take u)-11]
12 let a=zero?(0)
    in debug(0)";;
14 Environment:
(a, BoolVal true)
16 Module m2[(v, NumVal 33)]
Module m1[(u, NumVal 44)]
18 Store:
Empty
20 - : Ds.exp_val ReM.result = ReM.Ok Ds.UnitVal
utop

```

6.3 Type-Checking

6.3.1 Specification

Judgements for typing programs	$\vdash \text{AProg}(ms, e) : t$
Judgements for typing expressions	$\Delta; \Gamma \vdash e : t$
Judgements for typing list of module declarations	$\Delta_1 \vdash ms : \Delta_2$

Γ is the standard type environment from before Δ is a module type environment and is required for typing the expression `from m take x`

A **module type** is an expression of the form $m[u_1 : t_1, \dots, u_n : t_n]$. A **module type environment** is a sequence of module types.

$$\Delta ::= \epsilon \mid m[u_1 : t_1, \dots, u_n : t_n] \Delta$$

We use letters Δ to denote module type environments. The empty module type environment is written ϵ . If $m[u_1 : t_1, \dots, u_n : t_n] \in \Delta$, then we $m \in \text{dom}(\Delta)$. Moreover, in that case, have $u_i \in \text{dom}(\Delta(m))$, for $i \in 1..n$, and also $\Delta(m, u_i) = t_i$.

There is just one typing rule for typing programs, namely TProg. There is one new typing rule for expressions, namely TFromTake, it allows to type qualified variables. There are two typing rules for lists of module definitions: one for when the list is empty (TModE) and one for when it is not (TModNE). Regarding the latter,

$$\begin{array}{c}
\frac{\epsilon \vdash_{\text{ms}} :: \Delta \quad \Delta; \epsilon \vdash e :: t}{\vdash \text{AProg}(\text{ms}, e) :: t} \text{TProg} \\
\\
\frac{\text{m} \in \text{dom}(\Delta) \quad x \in \text{dom}(\Delta(\text{m})) \quad \Delta(\text{m}, x) = t}{\Delta; \Gamma \vdash \text{from m take x} :: t} \text{TFromTake} \\
\\
\frac{}{\Delta \vdash \epsilon :: \epsilon} \text{TModE} \\
\\
\frac{\begin{array}{c} [x_i]_{i \in I} \triangleleft [y_j]_{j \in J} \\ (\Delta_1; [y_1 := s_1] \dots [y_{j-1} := s_{j-1}] \Gamma \vdash e_j :: s_j)_{j \in J} \\ (t_i = s_{f(i)})_{i \in I} \\ \text{m}[x_i : t_i]_{i \in I} \Delta_1 \vdash_{\text{ms}} :: \Delta_2 \end{array}}{\Delta_1 \vdash \text{m}[x_i : t_i]_{i \in I} [y_j = e_j]_{j \in J} \text{ms} :: \text{m}[x_i : t_i]_{i \in I} \Delta_2} \text{TModNE}
\end{array}$$

Figure 6.2: Typing rules for SIMPLE-MODULES

- Δ_2 is the type of the list of modules ms
- $\text{m}[x_i : t_i]_{i \in I} \Delta_1$ is the type of the list of modules that ms can use
- $[x_i]_{i \in I} \triangleleft [y_j]_{j \in J}$ means that the list of variables $[x_i]_{i \in I}$ is a sublist of the list of variables $[y_j]_{j \in J}$. This relation determines an injective, order preserving function $f : I \rightarrow J$

6.3.2 Implementation

```

type tenv =
2 | EmptyTEnv
  | ExtendTEnv of string*texpr*tenv
4 | ExtendTEnvMod of string*tenv*tenv
dst.ml

let rec
2 type_of_prog (AProg (ms,e)) =
  type_of_modules ms >>+
4 chk_expr e
and
6 type_of_modules : module_decl list -> tenv tea_result =
  fun mdecls ->
8 List.fold_left
  (fun curr_tenv (AModDecl(mname,ASimpleInterface(expected_iface),mbody)) ->
10 curr_tenv >>+
  (type_of_module_body mbody >>= fun i_body ->
12 if (is_subtype i_body expected_iface)
  then
14 extend_tenv_mod mname (var_decls_to_tenv expected_iface)
  else
16 error("Subtype failure: "^mname))
  )

```

```
18     lookup_tenv
19     mdecls
20 and
21   type_of_module_body : module_body -> tenv tea_result =
22   fun (AModBody vdefs) ->
23     lookup_tenv >>= fun glo_tenv ->
24     (List.fold_left (fun loc_tenv (var,decl) ->
25       loc_tenv >>+
26       (append_tenv_rev glo_tenv >>+
27        chk_expr decl >>=
28        extend_tenv var))
29       (empty_tenv ()))
30     vdefs) >>= fun tmbody ->
31     return (reverse_tenv tmbody)
32 and
33   chk_expr : expr -> texpr tea_result =
34   fun e ->
35     match e with
36     | Int n -> return IntType
37     | Var id -> apply_tenv id
38     | QualVar(module_id,var_id) ->
39       apply_tenv_qual module_id var_id
40     ...
```

6.4 Further Reading

Module inclusion Private types First-class modules

Appendix A

Supporting Files

A.1 File Structure

Typical file structure for an interpreter (in this example, ARITH).

```
.
|____arith.opam
|____dune-project
|____build
|____lib
| |____.merlin
| |____.ocamlinit
| |____ds.ml
| |____dune
| |____interp.ml
|____test
| |____dune
| |____test.ml
```

The source files are in the `lib` directory and the unit tests are in the `test` directory.

<code>.merlin</code>	Tells Merlin where to locate sources
<code>.ocamlinit</code>	Loaded by utop upon execution; opens some modules
<code>ds.ml</code>	Supporting data structures including expressed values, environments and results
<code>dune</code>	Configuration file for dune
<code>interp.ml</code>	Interpreter
<code>test.ml</code>	Unit tests

File structure for the parser.

```
.
|____parser_plaf.opam
|____dune-project
|____build
```

```
|_lib
| |_ast.ml
| |_grammar.mly
| |_lexer.mly
| |_dune
| |_parser.ml
| |_parserMessages.messages
|_test
| |_dune
| |_test.ml
```

The source files are in the `lib` directory and the unit tests are in the `test` directory.

<code>ast.ml</code>	Abstract Syntax
<code>dune</code>	Configuration file for dune
<code>lexer.mly</code>	Lexer generator
<code>grammar.mly</code>	Grammar for parser generator
<code>parserMessages.messages</code>	Error messages for parser (under construction)
<code>parser.ml</code>	Parser
<code>test.ml</code>	Unit tests

We use the `dune` build system for OCaml. You can find documentation on `dune` at <https://readthedocs.org/projects/dune/downloads/pdf/latest/>. Some common `dune` commands:

- Build the project and then run utop

```
$ dune utop
```

- Build the project

```
$ dune build
```

- Clean the current project (erasing `_build` directory)

```
$ dune clean
```

- Run tests (building if necessary)

```
$ dune runtest
```

- Generate documentation (install `odoc` with `opam` first):

```
$ dune build @doc
```

The generated html files are in:

```
$ open _build/default/_doc/_html/index.html
```

A.2 Running the Interpreters

First you must build and install the parser.

```
1 # cd PLaF/src/parser_plaf
  # dune build
3 Read 203 sample input sentences and 203 error messages.
  Read 203 sample input sentences and 203 error messages.
5 Read 203 sample input sentences and 203 error messages.
  # dune install parser_plaf
```

bash

Building and running the LET interpreter:

```
# cd PLaF/src/let/lib
2 # dune utop
```

bash

This builds and runs utop. You should be able to parse a simple expression by typing:

```
utop # parse "3+3";;
2 - : prog = AProg ([], Add (Int 3, Int 3))
```

utop

You may also evaluate an expression as follows:

```
# interp "let x=2 in x+x";;
2 - : exp_val Let.Ds.result = Ok (NumVal 4)
```

utop

Appendix B

Solution to Selected Exercises

Section 2.1

Answer B.0.1 (Exercise 2.2.1). *Sample expressions of each of the following types are:*

1. *expr*. An example is: `Int 2`.
2. *env*. Examples are: `EmptyEnv` and `ExtendEnv("x", NumVal 2, EmptyEnv)`
3. *exp_val*. Examples are: `NumVal 3` and `BoolVal true`.
4. *exp_val result*. Examples are: `Ok (NumVal 3)` and `Ok (BoolVal true)` and `Error "oops"`.
5. *int result*. Examples are: `Ok 1` and `Error "oops"`.
6. *env result*. Examples are: `Ok EmptyEnv` and `Ok (ExtendEnv("x", NumVal 2, EmptyEnv))`.
7. *int ea_result*. Examples are: `return 2` and `error "oops"`.
8. *exp_val ea_result*. Examples are: `return (NumVal 2)` and `return (BoolVal true)` and `error "oops"`.
Also, `apply_env "x"`.
9. *env ea_result*. Examples are: `return (EmptyEnv)` and `error "oops"`. Also, `extend_env "x" (NumVal 7)`.

Section 2.3

Answer B.0.2 (Exercise 2.4.8).

```
1 let even = proc (e) { proc (o) { proc (x) {  
2   if zero?(x)  
3   then zero?(0)  
4   else (((o e) o) (x-1)) }}}  
5 in  
6 let odd = proc(e) { proc (o) { proc (x) {  
7   if zero?(x)  
8   then zero?(1)  
9   else (((e e) o) (x-1)) }}}  
10 in (((even even) odd) 4)
```

Answer B.0.3 (Exercise 2.4.9).

```

1  let f =
2    proc (g) {
3      proc (leaf) {
4        proc (depth) {
5          if zero?(depth) then (leaf, leaf)
6          else (((g g) leaf) (depth-1)), (((g g) leaf) (depth-1))) }}}
7  in let pbt = proc (leaf) { proc (height) { ((f f) leaf) height) }}
8  in ((pbt 2) 3)

```

Section 2.5**Answer B.0.4** (Exercise 2.5.2).

```

1  # interp "
2  let l1 = cons(1, cons(2, cons(3, emptylist)))
3  in let l2 = cons(4, cons(5, emptylist))
4  in letrec append(l1) = proc (l2) {
5      if empty?(l1)
6      then l2
7      else cons(hd(l1), ((append tl(l1)) l2))
8  }
9  in ((append l1) l2)
10 ";
11 - : exp_val Rec.Ds.result =
12 Ok (ListVal [NumVal 1; NumVal 2; NumVal 3; NumVal 4; NumVal 5])
13
14 # interp "
15 let l = cons(1, cons(2, cons(3, emptylist)))
16 in let succ = proc (x) { x+1 }
17 in letrec map(l) = proc (f) {
18     if empty?(l)
19     then emptylist
20     else cons((f hd(l)), ((map tl(l)) f))
21 }
22 in ((map l) succ)
23 ";
24 - : exp_val Rec.Ds.result = Ok (ListVal [NumVal 2; NumVal 3; NumVal
25 4])
26
27 # interp "
28 let l = cons(1, cons(2, cons(1, emptylist)))
29 in let is_one = proc (x) { zero?(x-1) }
30 in letrec filter(l) = proc (p) {
31     if empty?(l)
32     then emptylist
33     else (if (p hd(l))
34         then cons(hd(l), ((filter tl(l)) p))
35         else ((filter tl(l)) p))
36 }
37 in ((filter l) is_one)";
38 - : exp_val Rec.Ds.result = Ok (ListVal [NumVal 1; NumVal 1])
39

```

utop

Answer B.0.5 (Exercise 2.5.3). 1. *debug* Must be placed in the *then* case:

```

let z = 0
2 in let prod = proc (x) { proc (y) { x*y }}
in letrec f(n) = if zero?(n) then debug(1) else ((prod n) (f (n-1)))
4 in (f 10)

```

2. Two possible solutions are:

```

let z = 0
2 in debug(let prod = proc (x) { proc (y) { x*y }})
in letrec f(n) = if zero?(n) then 1 else ((prod n) (f (n-1)))
4 in (f 10))

```

or:

```

let z = 0
2 in let prod = debug(proc (x) { proc (y) { x*y }})
in letrec f(n) = if zero?(n) then 1 else ((prod n) (f (n-1)))
4 in (f 10)

```

3. *debug* must be placed in the body of *proc* (x):

```

let z = 0
2 in let prod = proc (x) { debug(proc (y) { x*y })}
in letrec f(n) = if zero?(n) then 1 else ((prod n) (f (n-1)))
4 in (f 10)

```

Bibliography

[Fri08] Daniel P. Friedman. Essentials of Programming Languages. The MIT Press, 3rd edition, 2008.